

Epigram: Practical Programming with Dependent Types

Conor McBride

School of Computer Science and Information Technology
University of Nottingham

1 Motivation

Find the type error in the following Haskell expression:

```
if null xs then tail xs else xs
```

You can't, of course: this program is obviously nonsense unless you're a type-checker. The trouble is that only certain computations make sense if the `null xs` test is `True`, whilst others make sense if it is `False`. However, as far as the type system is concerned, the type of the `then` branch is the type of the `else` branch is the type of the entire conditional. Statically, the test is irrelevant. Which is odd, because if the test really were irrelevant, we wouldn't do it. Of course, `tail []` doesn't go wrong—well-typed programs don't go wrong—so we'd better pick a different word for the way they do go.

Abstraction and application, tupling and projection: these provide the 'software engineering' superstructure for programs, and our familiar type systems ensure that these operations are used compatibly. However, sooner or later, most programs inspect data and make a choice—at that point our familiar type systems fall silent. They simply can't talk about specific data. All this time, we thought our *programming* was strongly typed, when it was just our *software engineering*. In order to do better, we need a static language capable of expressing the significance of particular values in legitimizing some computations rather than others. We should not give up on programming.

James McKinna and I designed Epigram [27, 26] to support a way of programming which builds more of the intended meaning of functions and data into their types. Its *style* draws heavily from the Alf system [13, 21]; its *substance* from my to Randy Pollack's Lego system [20, 23] Epigram is in its infancy and its implementation is somewhat primitive. We certainly haven't got everything right, nor have we yet implemented the whole design. We hope we've got *something* right. In these notes, I hope to demonstrate that such nonsense as we have seen above is not inevitable in real life, and that the extra articulacy which dependent types offer is both useful and *usable*. In doing so, I seek to stretch your imaginations towards what programming can be if we choose to make it so.

1.1 What are Dependent Types?

Dependent type systems, invented by Per Martin-Löf [22] generalize the usual function types $S \rightarrow T$ to dependent function types $\forall x : S \Rightarrow T$, where T may

mention—hence *depend* on— x . We still write $S \rightarrow T$ when T doesn't depend on x . For example, matrix multiplication may be typed¹

$$\mathbf{mult} : \forall i, j, k : \mathbf{Nat} \Rightarrow \mathbf{Matrix} \ i \ j \rightarrow \mathbf{Matrix} \ j \ k \rightarrow \mathbf{Matrix} \ i \ k$$

Datatypes like $\mathbf{Matrix} \ i \ j$ may depend on values fixing some particular property of their elements—a natural number indicating size is but one example. A function can specialize its return type to suit each argument. The typing rules for abstraction and application show how:

$$\frac{x : S \vdash t : T}{\lambda x \Rightarrow t : \forall x : S \Rightarrow T} \quad \frac{f : \forall x : S \Rightarrow T \quad s : S}{f \ s : [s/x]T}$$

Correspondingly, $\mathbf{mult} \ 2 \ 3 \ 1 : \mathbf{Matrix} \ 2 \ 3 \rightarrow \mathbf{Matrix} \ 3 \ 1 \rightarrow \mathbf{Matrix} \ 2 \ 1$ is the specialized multiplier for matrices of the given sizes.

We're used to universal quantification expressing *polymorphism*, but the quantification is usually over types. Now we can quantify over all values, and these include the types, which are values in \star . Our \forall captures many forms of abstraction uniformly. We can also see $\forall x : S \Rightarrow T$ as a logical formula and its inhabitants as a function which computes a *proof* of $[s/x]T$ given a particular value s in S . It's this correspondence between programs and proofs, the *Curry-Howard Isomorphism*, with the slogan 'Propositions-as-Types', which makes dependent type systems particularly suitable for representing computational logics.

However, if you want to check dependent types, be careful! Look again at the application rule; watch s hopping over the copula,² from the term side in the argument hypothesis (eg., our specific dimensions) to the type side in the conclusion (eg., our specific matrix types). With expressions in types, we must think again about when types are equal. Good old syntactic equality won't do: $\mathbf{mult} \ (1+1)$ should have the same type as $\mathbf{mult} \ 2$, so $\mathbf{Matrix} \ (1+1) \ 1$ should be the same type as $\mathbf{Matrix} \ 2 \ 1$! If we want computation to preserve types, we need at least to identify types with the same *normal forms*. Typechecking requires the *evaluation* of previously typechecked expressions—the phase distinction is still there, but it's slipperier.

What I like about dependent types is their precise language of data structures. In Haskell, we could define a sequence of types for lists of fixed lengths

```
data List0 x = Nil
data List1 x = Cons0 x (List0 x)
data List2 x = Cons1 x (List1 x)
```

but we'd have to stop sooner or later, and we'd have difficulty abstracting over either the whole collection, or specific subcollections like lists of even length. In

¹ We may write $\forall x : X; y : Y \Rightarrow T$ for $\forall x : X \Rightarrow \forall y : Y \Rightarrow T$ and $\forall x_1, x_2 : X \Rightarrow T$ for $\forall x_1 : X; x_2 : X \Rightarrow T$. We may drop the type annotation where inferrable.

² By *copula*, I mean the ':' which in these notes is used to link a term to its typing; Haskell uses '::', and the bold use the set-theoretic ' \in '.

Epigram, we can express the lot in one go, giving us the family of *vector* types with indices from `Nat` representing length. `Nat` is just an ordinary datatype.

$$\begin{aligned} & \underline{\text{data}} \left(\frac{}{\text{Nat} : \star} \right) \quad \underline{\text{where}} \left(\frac{}{\text{zero} : \text{Nat}} \right) ; \left(\frac{n : \text{Nat}}{\text{suc } n : \text{Nat}} \right) \\ & \underline{\text{data}} \left(\frac{n : \text{Nat} ; X : \star}{\text{Vec } n X : \star} \right) \quad \underline{\text{where}} \left(\frac{}{\text{vnil} : \text{Vec zero } X} \right) \\ & \qquad \qquad \qquad \left(\frac{x : X ; xs : \text{Vec } n X}{\text{vcons } x xs : \text{Vec } (\text{suc } n) X} \right) \end{aligned}$$

Inductive families [15], like `Vec`, are collections of datatypes, defined mutually and systematically, indexed by other data. Now we can use the dependent function space to give the ‘tail’ function a type which prevents criminal behaviour:

$$\text{vtail} : \forall n : \text{Nat} \Rightarrow \forall X : \star \Rightarrow \text{Vec } (\text{suc } n) X \rightarrow \text{Vec } n X$$

For no n is `vtail n X vnil` well typed. Indexed types state properties of their data which functions can rely on. They are the building blocks of Epigram programming. Our `Matrix i j` can just be defined as a vector of columns, say:

$$\underline{\text{let}} \left(\frac{\text{rows}, \text{cols} : \text{Nat}}{\text{Matrix } \text{rows } \text{cols} : \star} \right) \quad \text{Matrix } \text{rows } \text{cols} \Rightarrow \text{Vec } \text{cols } (\text{Vec } \text{rows } \text{Nat})$$

Already in Haskell, there are hooks available to crooks who want more control over data. One can exploit *non-uniform* polymorphism to enforce some kinds of structural invariant [31], like this

```
data Rect col x = Columns [col] | Longer (Rect (col, x))
type Rectangular = Rect ()
```

although this type merely enforces rectangularity, rather than a specific size. One can also collect into a `Vec` type class those functors which generate vector structures [24]. Matrix multiplication then acquires a type like

```
mult :: (Vec f, Vec g, Vec h) => f (g Int) -> g (h Int) -> f (h Int)
```

Programming with these ‘fake’ dependent types is an entertaining challenge, but let’s be clear: these techniques are cleverly dreadful, rather than dreadfully clever. Hideously complex dependent types certainly exist, but they express basic properties like size in a straightforward way—why should the length of a list be anything less ordinary than a *number*? In Epigram, it doesn’t matter whether the size of a matrix is statically determined or dynamically supplied—the size invariants are enforced, maintained and exploited, regardless of phase.

1.2 What is Epigram?

Epigram is a dependently typed functional programming language. On the surface, the system is an integrated editor-typechecker-interpreter for the language, owing a debt to the Alf [13, 21] and Agda [12] family of proof editors. Underneath, Epigram has a tactic-driven proof engine, like those of Coq [11] and Epigram’s immediate ancestor, the ‘Oleg’ variant of Lego [20, 23]. The latter has proof tactics which mimic Alf’s pattern matching style of proof in a more spartan type theory (Luo’s UTT [19]); James McKinna and I designed Epigram [27] as a ‘high-level programming’ interface to this technology. An Epigram program is really a tree of proof tactics which drive the underlying construction in UTT.

But this doesn’t answer the wider cultural question of what Epigram is. How does it relate to functional languages like SML [29] and Haskell [32]? How does it relate to previous dependently typed languages like DML [39] and Cayenne [4]? How does it relate pragmatically to more conventional ways of working in type theory in the systems mentioned above? What’s new?

I’ll return to these questions at the end of these notes, when I’ve established more of the basis for a technical comparison, but I can say this much now: DML refines the ML type system with numerical indexing, but the programs remain the same—erase the indices and you have an ML program; Cayenne programs are LazyML programs with a more generous type system, including programs at the type level, but severely restricted support for inductive families. Epigram is not an attempt to strap a more powerful type system to standard functional programming constructs—it’s rather an attempt to rethink what programming can become, given such a type system.

Dependent types can make explicit reference to programs and data. They can talk about *programming* in a way that simple types can’t. In particular, an *induction principle* is a dependent type. We learned this one as children:

$$\begin{aligned} \mathbf{NatInd} : \forall P : \mathbf{Nat} \rightarrow \star \Rightarrow \\ P \mathbf{zero} \rightarrow (\forall n : \mathbf{Nat} \Rightarrow P n \rightarrow P (\mathbf{suc} n)) \rightarrow \\ \forall n : \mathbf{Nat} \Rightarrow P n \end{aligned}$$

It gives rise to a proof technique—to give a proof of a more general proposition $P n$, give proofs that P holds for more specific *patterns* which n can take. Now cross out ‘proof’ and write ‘program’. The induction principle for \mathbf{Nat} specifies a particular strategy of case analysis and recursion, and Epigram can read it as such. Moreover, we can readily execute ‘proofs’ by induction, recursively applying the step program to the base value, to build a proof for any specific n :

$$\begin{aligned} \mathbf{NatInd} P \mathit{mz} \mathit{ms} \mathbf{zero} &\rightsquigarrow \mathit{mz} \\ \mathbf{NatInd} P \mathit{mz} \mathit{ms} (\mathbf{suc} n) &\rightsquigarrow \mathit{ms} n (\mathbf{NatInd} P \mathit{mz} \mathit{ms} n) \end{aligned}$$

Usually, functional languages have hard-wired constructs for constructor case analysis and general recursion; Epigram supports programming with *any* matching and recursion which you can specify as an induction principle and implement as a function. Epigram also supports a first-order method of implementing new induction principles—they too arise from inductive families.

It may surprise (if not comfort) functional programmers to learn that dependently typed programming seems odd to type theorists too. Type theory is usually seen either as the integration of ‘ordinary’ programming with a logical *superstructure*, or as a constructive logic which permits programs to be quietly extracted from proofs. Neither of these approaches really exploits dependent types in the programs and data themselves. At time of writing, neither Agda nor Coq offers substantial support for the kind of data structures and programs we shall develop in these notes, even though Alf and ‘Oleg’ did!

There is a tendency to see programming as a fixed notion, essentially untyped. In this view, we make sense of and organise programs by assigning types to them, the way a biologist classifies species, and in order to classify more the exotic creatures, like `printf` or the `zipWith` family, one requires more exotic types. This conception fails to engage with the full potential of types to make a positive contribution to program construction. Given what types can now express, let us open our minds afresh to the design of programming language *constructs*, and of programming *tools* and of the *programs* we choose to write anyway.

1.3 Overview of the Remaining Sections

- 2 Warm Up; Add Up** tries to give an impression of Epigram’s interactive style programming and the style of the programs via very simple examples—addition and the Fibonacci function. I expose the rôle of dependent types behind the scenes, even in simply typed programming.
- 3 Vectors and Finite Sets** introduces some very basic datatype families and operations—I explore `Vec` and also the family `Fin` of finite enumeration types, which can be used to index vectors. I show how case analysis for dependent types can be more powerful and more subtle than its simply typed counterpart.
- 4 Representing Syntax** illustrates the use of dependent types to enforce key invariants in expression syntax—in particular, the λ -calculus. I begin with untyped de Bruijn terms after the manner of Bird and Paterson [9] and end with simply typed de Bruijn terms in the manner of Altenkirch and Reus [2].³ On the way, I’ll examine some pragmatic issues in data structure design.
- 5 Is Looking Seeing?** homes in on the crux of dependently typed programming—*evidence*. Programs over indexed datatypes may *enforce* invariants, but how do we *establish* them? This section explores our approach to data analysis [27], expressing inspection as a form of induction and deriving induction principles for old types by defining new families.
- 6 Well Typed Programs which Don’t Go Wrong** shows the development of two larger examples—a typechecker for simply typed λ -calculus which yields a typed version *of its input* or an informative diagnostic, and a tagless and total evaluator for the well typed terms so computed.
- 7 Epilogue** reflects on the state of Epigram and its future in relation to what’s happening more widely in type theory and functional programming.

³ As I’m fond of pointing out, these papers were published almost simultaneously and have only one reference in common. I find that shocking!

I've dropped from these notes a more formal introduction to type theory: which introductory functional programming text explains how the typechecker works within the first forty pages? A precise understanding of type theory isn't necessary to engage with the ideas, get hold of the basics and start programming. I'll deal with technicalities as and when we encounter them. If you do feel the need to delve deeper into the background, there's plenty of useful literature out there—the next subsection gives a small selection.

1.4 Some Useful Reading

Scholars of functional programming and of type theory should rejoice that they now share much of the same ground. It would be terribly unfortunate for the two communities each to fail to appreciate the potential contribution of the other, through cultural ignorance. We must all complain less and read more!

For a formal presentation of the Epigram language, see '*The view from the left*' [27]. For a deeper exploration of its underlying type theory—see '*Computation and Reasoning: A Type Theory for Computer Science*' [19] by Zhaohui Luo. User documentation, examples and solutions to exercises are available online [26].

Much of the impetus for Epigram comes from proof assistants. Proof and programming are similar activities, but the tools have a different feel. I can recommend '*Coq'Art*' [7] by Yves Bertot and Pierre Castéran as an excellent tutorial for this way of working, and for the Coq system in particular. The tactics of a theorem prover animate the rules of its underlying type theory, so this book also serves as a good practical introduction to the more formal aspects.

The seminal textbook on type theory as a programming language is '*Programming in Martin-Löf's type theory: an introduction*' [35] by Bengt Nordström, Kent Petersson and Jan Smith. It is now fifteen years old and readily available electronically, so there's no excuse to consider type theory a closed book.

Type theorists should get reading too! Modern functional programming uses richer type systems to express more of the structure of data and capture more patterns of computation. I learned a great deal from '*Algebra of Programming*' by Richard Bird and Oege de Moor [8]. It's a splendid and eye-opening introduction to a more categorical and calculational style of correct program construction. '*Purely Functional Data Structures*' by Chris Okasaki [30] is a delightful compendium of data structures and algorithms, clearly showing the advantages of fitting datatypes more closely to algorithms.

The literature on overloading, generic programming, monads, arrows, higher-order polymorphism is too rich to enumerate, but it raises important issues which type theorists must address if we want to make a useful contribution to functional programming in practice. I'd advise hungry readers to start with this very series of Advanced Functional Programming lecture notes.

1.5 For Those of you Watching in Black & White

Before we start in earnest, let's establish typographical conventions and relate the system's display with these notes. Epigram's syntax is two-dimensional: the

buffer contains a *document* with a rectangular region selected—highlighted with a bright background. A *document* is a vertical sequence of lines; a *line* is a horizontal sequence of boxes; a *box* is either a character, or a bracket containing a document. A *bracket* is either a

```
(      !
group, !      ···      ! , which has the usual functions of parenthesis, or a
      !      )
```

```
[      !
shed, !      ···      ! , where you can tinker with text as you please. An Epigram
      !      ]
```

line may thus occupy more than one ASCII line. If a bracket is opened on a physical line, it must either be closed on that line or *suspended* with a `!`, then *resumed* on the next physical line with another `!`. I hasten to add that the Epigram editor does all of this box-drawing for you. You can fit two Epigram lines onto one physical line by separating them with `;`, and split one Epigram line into two physical lines by prefixing the second with `%`.

The Epigram document is a syntax tree in which leaves may be sheds—their contents are monochrome and belong to you. You can edit any shed without Epigram spying on you, but the rest of the document gets *elaborated*—managed, typechecked, translated to UTT, typeset, coloured in and generally abused by the system. In particular, Epigram colours recognized identifiers. There is only one namespace—this colour is just for show. If you can't see the colours in your copy of these notes, don't worry: I've adopted font and case conventions instead.

Blue	sans serif, uppercase initial	type constructor
red	sans serif, lowercase initial	data constructor
green	serif, boldface	defined variable
<i>purple</i>	serif, italic	abstracted variable
<u>black</u>	serif, underlined	reserved word

These conventions began in my handwritten slides—the colour choices are more or less an accident of the pens available, but they seem to have stuck. Epigram also has a convention for background colour, indicating the elaboration status of a block of source code.

white (light green when selected) indicates successful elaboration

yellow indicates that Epigram cannot yet see why a piece of code is good

brown indicates that Epigram can see why a piece of code is bad

Yellow backgrounds come about when typing constraints cannot yet be solved, but it's still possible for the variables they involve to become more instantiated, allowing for a solution in the future.

There are some pieces of ASCII syntax which I cannot bring myself to uglify in L^AT_EX. I give here the translation table for tokens and for the extensible delimiters of two-dimensional syntax:

*	∀	λ	→	∧	⇒	⇐	()	[]	---
*	all	lam	->	/∧	=>	<=	(!)	[!]	---

Moreover, to save space here, I adopt an end-of-line style with braces {}, where the system puts them at the beginning.

At the top level, the document is a vertical sequence of **declarations** delineated by *rules*. A *rule* is a sequence of at least three ---. The initial document has just one declaration, consisting of a shed, waiting for you to start work.

1.6 Acknowledgements

I'd like to thank the editors, Tarmo Uustalu and Varmo Vene, and the anonymous referees, for their patience and guidance. I'd also like to thank my colleagues and friends, especially James McKinna, Thorsten Altenkirch, Zhaohui Luo, Paul Callaghan, Randy Pollack, Peter Hancock, Edwin Brady, James Chapman and Peter Morris. Sebastian Hanowski and Wouter Swierstra deserve special credit for the feedback they have given me. Finally, to those who were there in Tartu, thank you for making the experience one I shall always value.

This work was supported by EPSRC grants GR/R72259 and EP/C512022.

2 Warm Up; Add Up

Let's examine the new technology in the context of a simple and familiar problem: adding natural numbers. We have seen this Epigram definition:⁴

$$\underline{\text{data}} \left(\frac{\quad}{\text{Nat} : \star} \right) \underline{\text{where}} \left(\frac{\quad}{\text{zero} : \text{Nat}} \right) ; \left(\frac{n : \text{Nat}}{\text{suc } n : \text{Nat}} \right)$$

We can equally well define the natural numbers in Epigram as follows:

$$\underline{\text{data}} \text{ Nat} : \star \underline{\text{where}} \text{ zero} : \text{Nat} ; \text{ suc} : \text{Nat} \rightarrow \text{Nat}$$

In Haskell, we would write 'data Nat = Zero | Suc Nat'.

The 'two-dimensional' version is a first-order presentation in the style of natural deduction rules⁵ [34]. Above the line go hypotheses typing the arguments; below, the conclusion typing a template for a value. The declarations 'zero is a Nat; if n is a Nat, then so is suc n' tell us what Nats look like. We get the actual types of zero and suc implicitly, by discharging the hypotheses.

⁴ Unary numbers are not an essential design feature; rest assured that primitive binary numbers will be provided eventually [10].

⁵ Isn't this a big step backwards for brevity? Or does Haskell's brevity depend on some implicit presumptions about what datatypes can possibly be?

We may similarly declare our function in the natural deduction style:

$$\underline{\text{let}} \left(\frac{x, y : \text{Nat}}{\text{plus } x \ y : \text{Nat}} \right)$$

This signals our intention to define a function called **plus** which takes two natural numbers and returns a natural number. The machine responds

plus $x \ y$ []

by way of asking ‘So **plus** has two arguments, x and y . What should it do with them?’. Our type signature has become a *programming problem* to be solved interactively. The machine supplies a **left-hand side**, consisting of a function symbol applied to **patterns**, binding **pattern variables**—initially, the left-hand side looks just like the typical application of **plus** which we declared and the pattern variables have the corresponding names. But they are binding occurrences, not references to the hypotheses. The scope of a rule’s hypotheses extends only to its conclusion; each problem and subproblem has its own scope.

Sheds [] are where we develop solutions. The basic editing operation in Epigram is to expose the contents of a shed to the elaborator. We can write the whole program in one shed, then elaborate it; or we can work a little at a time. If we select a shed, Epigram will tell us what’s in scope (here, x, y in **Nat**), and what we’re supposed to be doing with it (here, explaining how to compute **plus** $x \ y$ in **Nat**). We may proceed by filling in a **right-hand side**, explaining how to reduce the current problem to zero or more subproblems.

I suggest that we seek to define **plus** by structural recursion on x , entering the right-hand side $\Leftarrow \underline{\text{rec}} x$. The ‘ \Leftarrow ’ is pronounced ‘by’: it introduces right-hand sides which explain *by what means* to reduce the problem. Here we get

plus $x \ y \Leftarrow \underline{\text{rec}} x \{$
plus $x \ y$ [] $\}$

Apparently, nothing has changed. There is no presumption that recursion on x will be accompanied immediately (or ever) by case analysis on x . If you select the shed, you’ll see that something has changed—the context of the problem has acquired an extra hypothesis, called a *memo-structure*. A precise explanation must wait, but the meaning of the problem is now ‘construct **plus** $x \ y$, given x, y and the ability to call to **plus** on structural subterms of x ’. So let us now analyse x , proceeding $\Leftarrow \underline{\text{case}} x$.

plus $x \ y \Leftarrow \underline{\text{rec}} x \{$
plus $x \ y \Leftarrow \underline{\text{case}} x \{$
plus $\text{zero } y$ []
plus $(\text{suc } x) \ y$ [] $\}\}$

The two subproblems are precisely those corresponding to the two ways x could have been made by constructors of **Nat**. We can certainly finish the first

one off, entering $\Rightarrow y$. (The ‘ \Rightarrow ’ is ‘return’.) We can make some progress on the second by deciding to return the successor of something, $\Rightarrow \text{suc } []$.

$$\begin{aligned} \text{plus } x y &\Leftarrow \text{rec } x \{ \\ &\text{plus } x y \Leftarrow \text{case } x \{ \\ &\text{plus zero } y \Rightarrow y \\ &\text{plus (suc } x) y \Rightarrow \text{suc } [] \} \} \end{aligned}$$

Select the remaining shed and you’ll see that we have to fill in an element of **Nat**, given x, y and a subtly different memo-structure. Case analysis has instantiated the original argument, so we now have the ‘ability to make recursive calls to **plus** on structural subterms of $(\text{suc } x)$ ’ which amounts to the more concrete and more useful ‘ability to make recursive calls to **plus** on structural subterms of x , and on x itself’. Good! We can finish off as follows:

$$\begin{aligned} \text{plus } x y &\Leftarrow \text{rec } x \{ \\ \text{plus } x y &\Leftarrow \text{case } x \{ \\ \text{plus zero } y &\Rightarrow y \\ \text{plus (suc } x) y &\Rightarrow \text{suc (plus } x y) \} \} \end{aligned}$$

2.1 Who did the Work?

Two pages to add unary numbers? And that’s a simple example? If it’s that much like hard work, do we really want to know? Well, let’s look at how much was work and how much was culture shock. We wrote the bits in the boxes:

$$\text{let } \left(\frac{x, y : \text{Nat}}{\text{plus } x y : \text{Nat}} \right) ; \text{plus } x y \boxed{\Leftarrow \text{rec } x} \{ \\ \text{plus } x y \boxed{\Leftarrow \text{case } x} \{ \\ \text{plus zero } y \boxed{\Rightarrow y} \\ \text{plus (suc } x) y \boxed{\Rightarrow \text{suc (plus } x y)} \} \}$$

We wrote the type signature: we might have done that for virtue’s sake, but virtue doesn’t pay the rent. Here, we were repaid—we exchanged the usual hand-written left-hand sides for machine-generated patterns resulting from the conceptual step (usually present, seldom written) $\Leftarrow \text{case } x$. This was only possible because the machine already knew the type of x . We also wrote the $\Leftarrow \text{rec } x$, a real departure from conventional practice, but we got repaid for that too—we (humans and machines) know that **plus** is *total*.

Perhaps it’s odd that the program’s text is not entirely the programmer’s work. It’s the record of a partnership where we say what the plan is and the machine helps us carry it out. Contrast this with the ‘type inference’ model of programming, where we write down the details of the execution and the machine tries to guess the plan. In its pure form, this necessitates the restriction of plans to those which are blatant enough to be guessed. As we move beyond the Hindley-Milner system, we find ourselves writing down type information anyway.

‘Type inference’ thus has two aspects: ‘top-level inference’—inferring type schemes, as with Hindley-Milner ‘let’—and ‘program inference given types’—inferring details when a scheme is instantiated, as with Hindley-Milner variables. Epigram rejects the former, but takes the latter further than ever. As types represent a higher-level design statement than programs, we should prefer to write types if they make programs cheaper.

Despite its interactive mode of construction, Epigram is fully compliant with the convention that a file of source code, however manufactured, contains all that’s required for its recognition as a program. The bare text, without colour or other markup, is what gets elaborated. The elaboration process for a large code fragment just reconstructs a suitable interactive development offstage, cued by the program text—this is how we reload programs. You are free to negotiate your own compromise between incremental and batch-mode programming.

2.2 Where are the Dependent Types?

The type of **plus** is unremarkably simple, but if you were watching closely, you’ll have noticed that the *machine* was using dependent types the whole time. Let’s take a closer look. Firstly, a thought experiment—define a primitive recursion operator for `Nat` in Haskell as follows:

```
primRec{p} :: Nat -> p -> (Nat -> p -> p) -> p
primRec{p} Zero    mz ms = mz
primRec{p} (Suc n) mz ms = ms n (primRec{p} n mz ms)
```

I’ve made `primRec`’s type parameter explicit in a comment so we can follow what happens as we put it to work. How might we write **plus**? Try applying `primRec` to the first argument, then checking what’s left to do:

```
plus :: Nat -> Nat -> Nat
plus = \ x -> primRec{Nat -> Nat} x mz ms where
  mz :: Nat -> Nat           -- fill this in
  ms :: Nat -> (Nat -> Nat) -> Nat -> Nat -- fill this in
```

Now we must fill in the *methods* `mz` and `ms`, but do their types show what rôle they play? There are seven occurrences⁶ of `Nat` in their types—which is which? Perhaps *you* can tell, because you understand `primRec`, but how would a machine guess? And if we were defining a more complex function this way, we might easily get lost—try defining equality for lists using `foldr`.

However, recall our **NatInd** principle with operational behaviour just like `primRec` but a type which makes clear the relationship between the methods and the patterns for which they apply. If we’re careful, we can use that extra information to light our way. Where `primRec` takes a constant type parameter, **NatInd** takes a *function* $P : \text{Nat} \rightarrow \star$. If we take $P\ x \rightsquigarrow \text{Nat} \rightarrow \text{Nat}$, we

⁶ If I had chosen a first-order recursion, there would have been as few as four, but that would presume to fix the second argument through the course of the recursion.

get the `primRec` situation. How might we use P 's argument to our advantage? Internally, Epigram doesn't build `plus : Nat → Nat → Nat` but rather a *proof*

$$\diamond\text{plus} : \forall x, y : \text{Nat} \Rightarrow \langle \text{plus } x \ y : \text{Nat} \rangle$$

We can interpret $\langle \text{plus } x \ y : \text{Nat} \rangle$ as the property of x and y that '`plus` $x \ y$ is a *computable* element of `Nat`'. This type is equipped with a constructor which packs up values and a function which runs computations

$$\frac{n : \text{Nat}}{\text{return}\langle \text{plus } x \ y \rangle n : \langle \text{plus } x \ y : \text{Nat} \rangle} \quad \frac{c : \langle \text{plus } x \ y : \text{Nat} \rangle}{\text{call}\langle \text{plus } x \ y \rangle c : \text{Nat}}$$

such that

$$\text{call}\langle \text{plus } x \ y \rangle (\text{return}\langle \text{plus } x \ y \rangle n) \rightsquigarrow n$$

Given $\diamond\text{plus}$, we may readily extract `plus`—apply and run!

$$\text{plus} \rightsquigarrow \lambda x, y \Rightarrow \text{call}\langle \text{plus } x \ y \rangle (\diamond\text{plus } x \ y) : \text{Nat} \rightarrow \text{Nat} \rightarrow \text{Nat}$$

Now, let's build $\diamond\text{plus}$ as a proof by `NatInd`:

$$\begin{aligned} \diamond\text{plus} &\rightsquigarrow \text{NatInd } (\lambda x \Rightarrow \forall y : \text{Nat} \Rightarrow \langle \text{plus } x \ y : \text{Nat} \rangle) \text{mz ms} \\ \text{where } \text{mz} &: \forall y : \text{Nat} \Rightarrow \langle \text{plus zero } y : \text{Nat} \rangle \\ \text{ms} &: \forall x : \text{Nat} \Rightarrow (\forall y : \text{Nat} \Rightarrow \langle \text{plus } x \ y : \text{Nat} \rangle) \\ &\quad \rightarrow \forall y : \text{Nat} \Rightarrow \langle \text{plus (suc } x) \ y : \text{Nat} \rangle \end{aligned}$$

It's not hard to see how to generate the left-hand sides of the subproblems for each case—just read them off from their types! Likewise, it's not hard to see how to translate the right-hand sides we supplied into the proofs—pack them up with `return` $\langle \dots \rangle$ translate recursive calls via `call` $\langle \dots \rangle$:

$$\begin{aligned} \text{mz} &\rightsquigarrow \lambda y \Rightarrow \text{return}\langle \text{plus zero } y \rangle \ y \\ \text{ms} &\rightsquigarrow \lambda x \Rightarrow \lambda xhyp \Rightarrow \lambda y \Rightarrow \\ &\quad \text{return}\langle \text{plus (suc } x) \ y \rangle (\text{suc } (\text{call}\langle \text{plus } x \ y \rangle (xhyp \ y))) \end{aligned}$$

From this proof, you can read off both the high-level program and the its low-level operational behaviour in terms of primitive recursion. And that's basically how Epigram works! Dependent types aren't just the basis of the Epigram *language*—the *system* uses them to organise even simply typed programming.

2.3 What are `case` and `rec`?

In the `plus` we actually wrote, we didn't use induction—we used `case` x and `rec` x . These separate induction into its aspects of distinguishing constructors and of justifying recursive calls. The keywords `case` and `rec` cannot stand alone, but `case` e and `rec` e are meaningful whenever e belongs to a datatype—Epigram constructs their meaning from the structure of that datatype.

In our example, $x : \text{Nat}$, and Epigram give us

$$\underline{\text{case}}\ x : \forall P : \text{Nat} \rightarrow \star \Rightarrow \\ (P\ \text{zero}) \rightarrow (\forall x' : \text{Nat} \Rightarrow P(\text{suc}\ x')) \rightarrow P\ x$$

This is an induction principle instantiated at x with its inductive hypotheses chopped off: it just says, ‘to do P with x , show how to do P with each of these patterns’. The associated computational behaviour puts proof into practice:

$$\begin{aligned} (\underline{\text{case}}\ \text{zero})\ P\ mz\ ms &\rightsquigarrow mz \\ (\underline{\text{case}}\ (\text{suc}\ x))\ P\ mz\ ms &\rightsquigarrow ms\ x \end{aligned}$$

There is nothing special about $\underline{\text{case}}\ x$. When elaborating $\Leftarrow e$, it’s the *type* of e which specifies how to split a problem into subproblems. If, as above, we take

$$P \rightsquigarrow \lambda x \Rightarrow \forall y : \text{Nat} \Rightarrow \langle \text{plus}\ x\ y : \text{Nat} \rangle$$

then the types of mz and ms give us the split we saw when we wrote the program. What about $\underline{\text{rec}}\ x$?

$$\underline{\text{rec}}\ x : \forall P : \text{Nat} \rightarrow \star \Rightarrow \\ (\forall x : \text{Nat} \Rightarrow (\underline{\text{memo}}\ x)\ P \rightarrow P\ x) \rightarrow \\ P\ x$$

This says ‘if you want to do $P\ x$, show how to do it given access to P for everything structurally smaller than x ’. This $(\underline{\text{memo}}\ x)$ is another gadget generated by Epigram from the structure of x ’s type—it uses the power of computation in types to capture the notion of ‘structurally smaller’:

$$\begin{aligned} (\underline{\text{memo}}\ \text{zero})\ P &\rightsquigarrow \text{One} \\ (\underline{\text{memo}}\ (\text{suc}\ n))\ P &\rightsquigarrow (\underline{\text{memo}}\ n)\ P \wedge P\ n \end{aligned}$$

That is $(\underline{\text{memo}}\ x)\ P$ is the type of a big tuple which memoizes P for everything smaller than x . If we analyse x , the memo-structure computes,⁷ giving us the trivial tuple for the **zero** case, but for **(suc n)**, we gain access to $P\ n$. Let’s watch the memo-structure unfolding in the inevitable Fibonacci example.

$$\underline{\text{let}}\ \left(\frac{n : \text{Nat}}{\text{fib}\ n : \text{Nat}} \right) ; \text{fib}\ n \Leftarrow \underline{\text{rec}}\ n \{ \\ \text{fib}\ n \Leftarrow \underline{\text{case}}\ n \{ \\ \text{fib}\ \text{zero} \Rightarrow \text{zero} \\ \text{fib}\ (\text{suc}\ n) \Leftarrow \underline{\text{case}}\ n \{ \\ \text{fib}\ (\text{suc}\ \text{zero}) \Rightarrow \text{suc}\ \text{zero} \\ \text{fib}\ (\text{suc}\ (\text{suc}\ n)) \Rightarrow \boxed{[]}\ \}} \}$$

⁷ Following a suggestion by Thierry Coquand, Eduardo Giménez shows how to separate induction into $\underline{\text{case}}$ and $\underline{\text{rec}}$ in [17]. He presents memo-structures inductively, to justify the *syntactic* check employed by Coq’s **Fix** construct. The computational version is mine [23]; its unfolding memo-structures give you a *menu* of recursive calls.

If you select the remaining shed, you will see that the memo structure in the context has unfolded (modulo trivial algebra) to:

$$(\text{memo } n) (\lambda x \Rightarrow \langle \text{fib } x : \text{Nat} \rangle) \wedge \langle \text{fib } n : \text{Nat} \rangle \wedge \langle \text{fib } (\text{suc } n) : \text{Nat} \rangle$$

which is just as well, as we want to fill in **plus** (**fib** n) (**fib** (**suc** n)).

At this stage, the approach is more important than the details. The point is that programming with \Leftarrow imposes *no fixed notion* of case analysis or recursion. Epigram does not have ‘pattern matching’. Instead, \Leftarrow admits whatever notion of problem decomposition is specified by the type of the expression (the *eliminator*) which follows it. The value of the eliminator gives the operational semantics to the program built from the solutions to the subproblems.

Of course, Epigram equips every datatype with case and rec, giving us the usual notions of *constructor* case analysis *structural* recursion. But we are free to make our own eliminators, capturing more sophisticated analyses or more powerful forms of recursion. By talking about *patterns*, dependent types give us the opportunity to specify and implement new ways of programming.

2.4 Pause for Thought

Concretely, we have examined one datatype and two programs. Slightly more abstractly, we have seen the general shape of Epigram programs as **decision trees**. Each node has a left-hand side, stating a programming problem p , and a right-hand-side stating how to attack it. The leaves of the tree $p \Rightarrow t$ explain directly what value to return. The internal nodes $p \Leftarrow e$ use the type of the eliminator e as a recipe for reducing the problem statement to subproblem statements, and the value of e as a recipe for solving the whole problem, given the solutions to the subproblems. Every datatype is equipped with eliminators of form case x for constructor case analysis and rec x for structural recursion, allowing us to construct obviously total programs in a pattern matching style.

However, the \Leftarrow construct is a more general tool than the **case** construct of conventional languages. We answer Wadler’s question of how to combine data abstraction with notions of pattern matching [38] by making notions of pattern matching first-class values.

It’s reasonable to ask ‘Can’t I write ordinary programs in an ordinary way? Must I build decision trees?’. I’m afraid the answer, for now, is ‘yes’, but it’s just a matter of syntactic sugar. We’re used to prioritized lists of patterns with a ‘take the first match’ semantics [28]. Lennart Augustsson showed us how to compile these into trees of **case-on-variables** [3]. Programming in Epigram is like being Augustsson’s compiler—you choose the tree, and it shows you the patterns. The generalization lies in what may sit at the nodes. One could flatten those regions of the tree with nonempty \Leftarrow case x nodes and use Augustsson’s algorithm to recover a decision tree, leaving \Leftarrow explicit only at ‘peculiar’ nodes.

Of course, this still leaves us with explicit \Leftarrow rec x supporting structural recursion. Can we get rid of that? There are various methods of spotting safe recursive calls [17]; some even extend to tracking guardedness through mutual

definitions [1]. We could use these to infer obvious appeals to `rec`, leaving only sophisticated recursions explicit. Again, it's a question of work. Personally, I want to write functions which are seen to be total.

Some might complain 'What's the point of a programming language that isn't Turing complete?', but I ask in return, 'Do you demand *compulsory ignorance* of totality?'. Let's guarantee totality explicitly whenever we can [37]. It's also possible, contrary to popular nonsense, to have dependent types and general recursive programs, preserving decidable typechecking: the cheapest way to do this is to work under an *assumed* eliminator with type

$$\forall P : \star \Rightarrow (P \rightarrow P) \rightarrow P$$

to which only the run time system gives a computational behaviour; a less drastic way is to treat general recursion as an impure monadic effect.

But in any case, you might be surprised how little you need general recursion. Dependent types make more programs structurally recursive, because dependent types have more structure. Inductive families with inductive indices support recursion on the data itself and recursion on the indices. For example, first-order unification [36] becomes structurally recursive when you index terms by the number of variables over which they are constructed—solving a variable may blow up the terms, but it decreases this index [25].

2.5 Some Familiar Datatypes

Just in time for the first set of exercises, let's declare some standard equipment. We shall need `Bool`, which can be declared like so:

```
data Bool :  $\star$  where true, false : Bool
```

The standard `Maybe` type constructor is also useful:

```
data (  $\frac{X : \star}{\text{Maybe } X : \star}$  ) where (  $\frac{}{\text{nothing} : \text{Maybe } X}$  ) ; (  $\frac{x : X}{\text{just } x : \text{Maybe } X}$  )
```

Note that I didn't declare `X` in the rules for `nothing` and `just`. The hypotheses of a rule scope only over its conclusion, so it's not coming from the `Maybe` rule. Rather, in each rule Epigram can tell from the way `X` is used that it must be a type, and it silently generalizes the constructors, just the way the Hindley-Milner system generalizes definitions.

It's the natural deduction notation which triggers this generalization. We were able to define `Bool` without it because there was nothing to generalize. Without the rule to 'catch' the `X`, plain

```
nothing : Maybe X
```

wouldn't exactly be an error. The out-of-scope `X` is waiting to be explained by some *prior* definition: the `nothing` constructor would then be specific to that `X`.

Rule-induced generalization is also happening here, for polymorphic lists:

$$\underline{\text{data}} \left(\frac{X : \star}{\text{List } X : \star} \right) \quad \underline{\text{where}} \left(\frac{}{\text{nil} : \text{List } X} \right) ; \left(\frac{x : X ; xs : \text{List } X}{\text{cons } x \text{ } xs : \text{List } X} \right)$$

We also need the binary trees with N -labelled nodes and L -labelled leaves.

$$\underline{\text{data}} \left(\frac{N, L : \star}{\text{Tree } N \text{ } L : \star} \right) \quad \underline{\text{where}} \left(\frac{l : L}{\text{leaf } l : \text{Tree } N \text{ } L} \right) ; \left(\frac{n : N ; s, t : \text{Tree } N \text{ } L}{\text{node } n \text{ } s \text{ } t : \text{Tree } N \text{ } L} \right)$$

2.6 Exercises: Structural Merge-Sort

To get used to the system, and to programming with structural recursion, try these exercises, which only involve the simple types above

Exercise 1 (le) Define the ‘less-or-equal’ test:

$$\underline{\text{let}} \left(\frac{x, y : \text{Nat}}{\text{le } x \text{ } y : \text{Bool}} \right)$$

Does it matter which argument you do rec on?

Exercise 2 (cond) Define the conditional expression:

$$\underline{\text{let}} \left(\frac{b : \text{Bool} ; \text{then}, \text{else} : T}{\text{cond } b \text{ then else} : T} \right)$$

Exercise 3 (merge) Use the above to define the function which **merges** two lists, presumed already sorted into increasing order, into one sorted list containing the elements from both.

$$\underline{\text{let}} \left(\frac{xs, ys : \text{List Nat}}{\text{merge } xs \text{ } ys : \text{List Nat}} \right)$$

Is this function structurally recursive on just one of its arguments? Nested recs combine lexicographically.

Exercise 4 (flatten) Use **merge** to implement a function **flatten**ing a tree which may have numbers at the leaves, to produce a sorted list of those numbers. Ignore the node labels.

$$\underline{\text{let}} \left(\frac{t : \text{Tree } N \text{ } (\text{Maybe Nat})}{\text{flatten } t : \text{List Nat}} \right)$$

We can have a structurally recursive $O(n \log n)$ sorting algorithm if we can share out the elements of a list into a *balanced* tree, then **flatten** it.

Exercise 5 (insert) Implement the **insertion** of a number into a tree:

$$\underline{\text{let}} \left(\frac{n : \text{Nat} ; t : \text{Tree Bool (Maybe Nat)}}{\text{insert } n \ t : \text{Tree Bool (Maybe Nat)}} \right)$$

Maintain this balancing invariant throughout: in **(node true s t)**, *s* and *t* contain equally many numbers, whilst in **(node false s t)**, *s* contains exactly one more number than *t*.

Exercise 6 (share, sort) Implement

$$\underline{\text{let}} \left(\frac{ns : \text{List Nat}}{\text{share } ns : \text{Tree Bool (Maybe Nat)}} \right)$$

$$\underline{\text{let}} \left(\frac{ns : \text{List Nat}}{\text{sort } ns : \text{List Nat}} \right)$$

so that **sort** sorts its input in $O(n \log n)$ time.

3 Vectors and Finite Sets

Moving on to dependent data structures now, let's take a closer look at **Vec**:

$$\underline{\text{data}} \left(\frac{n : \text{Nat} ; X : \star}{\text{Vec } n \ X : \star} \right) \underline{\text{where}} \left(\frac{}{\text{vnil} : \text{Vec zero } X} \right)$$

$$\left(\frac{x : X ; xs : \text{Vec } n \ X}{\text{vcons } x \ xs : \text{Vec (suc } n) \ X} \right)$$

The generalization mechanism ensures that all the previously undeclared variables arising inside each deduction rule are silently quantified in the resulting type, with the *implicit* \forall -quantifier. Written out in full, we have declared

$$\underline{\text{data}} \quad \text{Vec} : \text{Nat} \rightarrow \star \rightarrow \star$$

$$\underline{\text{where}} \quad \text{vnil} : \forall X : \star \Rightarrow \text{Vec zero } X$$

$$\text{vcons} : \forall X : \star \Rightarrow \forall n : \text{Nat} \Rightarrow X \rightarrow \text{Vec } n \ X \rightarrow \text{Vec (suc } n) \ X$$

On usage, Epigram tries to infer arguments for expressions with implicitly quantified types, just the way the Hindley-Milner system specializes polymorphic things—by solving the equational constraints which arise in typechecking. However, Epigram needs and supports a ‘manual override’: the postfix `_` operator inhibits inference and makes an implicit function explicit, so

$$\text{vnil}__ : \forall X : \star \Rightarrow \text{Vec zero } X$$

$$\text{vnil_Nat} : \text{Vec zero Nat}$$

To save space, I often write overridden arguments as subscripts—eg., `vnilNat`.

Given this definition, let's start to write some simple programs:

$$\underline{\text{let}} \left(\frac{ys : \text{Vec}(\text{suc } m) Y}{\text{vhead } ys : Y} \right) ; \text{vhead } ys \text{ [} \leftarrow \text{ case } ys \text{]}$$

What happens when we elaborate? Well, consider which constructors can possibly have made ys . Certainly not vnil , unless $\text{zero} = \text{suc } n$. We just get a vcons case—the one case we want, for ‘head’ and ‘tail’:

$$\underline{\text{let}} \left(\frac{ys : \text{Vec}(\text{suc } m) Y}{\text{vhead } ys : Y} \right) ; \text{vhead } ys \leftarrow \underline{\text{case}} \text{ } ys \{ \\ \text{vhead } (\text{vcons } y \text{ } ys) \Rightarrow y \}$$

$$\underline{\text{let}} \left(\frac{ys : \text{Vec}(\text{suc } m) Y}{\text{vtail } ys : \text{Vec } m Y} \right) ; \text{vtail } ys \leftarrow \underline{\text{case}} \text{ } ys \{ \\ \text{vtail } (\text{vcons } y \text{ } ys) \Rightarrow ys \}$$

In the latter, not only do we get that it's vcons as opposed to vnil : it's the particular vcons which extends vectors of the length we need. What's going on? Much as Thierry Coquand proposed in [13], Epigram is *unifying* the scrutinee of the case with the possible constructor patterns, in both term and type:

$ys : \text{Vec}(\text{suc } m) Y$	unifier
$\text{vnil}_X : \text{Vec } \text{zero } X$	impossible
$\text{vcons}_{X \ n} x \text{ } xs' : \text{Vec}(\text{suc } n) X$	$X = Y, n = m, xs = \text{vcons}_Y \ m \ x \ xs'$

Only the vcons case survives—Epigram then tries to choose names for the pattern variables which maintain a ‘family resemblance’ to the scrutinee, hence the $(\text{vcons } y \text{ } ys)$ in the patterns.

This unification doesn't just rule cases in or out: it can also feed information to type-level computations. Here's how to append vectors:

$$\underline{\text{let}} \left(\frac{xs : \text{Vec } m \ X ; \ ys : \text{Vec } n \ X}{\text{vappend}_m \ xs \ ys : \text{Vec}(\text{plus } m \ n) \ X} \right)$$

$$\text{vappend}_m \ xs \ ys \leftarrow \underline{\text{rec}} \text{ } xs \{ \\ \text{vappend}_m \ xs \ ys \leftarrow \underline{\text{case}} \text{ } xs \{ \\ \text{vappend}_{\text{zero}} \ \text{vnil} \ ys \Rightarrow ys \\ \text{vappend}_{(\text{suc } m)} (\text{vcons } x \text{ } xs) \ ys \Rightarrow \text{vcons } x \ (\text{vappend}_m \ xs \ ys) \}} \}$$

I've overridden the length arguments just to show what's happening—you can leave them implicit if you like. The point is that by looking at the first vector, we learn about its length. This lets plus compute exactly as we need for $ys : \text{Vec}(\text{plus } \text{zero } n) X$ in the vnil case. For vcons , the return type is $\text{Vec}(\text{plus}(\text{suc } m) n) X \rightsquigarrow \text{Vec}(\text{suc}(\text{plus } m \ n)) X$, which is what we supply.

3.1 Finite Sets

Let's examine the consequences of dependent case analysis for a different family:

$$\underline{\text{data}} \left(\frac{n : \text{Nat}}{\text{Fin } n : \star} \right) \underline{\text{where}} \left(\frac{}{\text{fz} : \text{Fin} (\text{suc } n)} \right) ; \left(\frac{i : \text{Fin } n}{\text{fs } i : \text{Fin} (\text{suc } n)} \right)$$

What happens when we elaborate this?

$$\underline{\text{let}} \left(\frac{i : \text{Fin zero}}{\text{magic } i : X} \right) ; \text{magic } i \text{ [} \leftarrow \text{ case } i \text{]}$$

You've probably guessed, but let's just check:

$i : \text{Fin zero}$	unifier
$\text{fz}_n : \text{Fin} (\text{suc } n)$	impossible
$\text{fs}_n j : \text{Fin} (\text{suc } n)$	impossible

So the finished product is just $\text{magic } i \leftarrow \underline{\text{case}} i$

The idea is that $\text{Fin } n$ is an enumeration type containing n values. Let's tabulate the first few members of the family, just to see what's going on. (I'll show the implicit arguments as subscripts, but write in decimal to save space.)

Fin 0	Fin 1	Fin 2	Fin 3	Fin 4	...
	fz ₀	fz ₁	fz ₂	fz ₃	...
		fs ₁ fz ₀	fs ₂ fz ₁	fs ₃ fz ₂	...
			fs ₂ (fs ₁ fz ₀)	fs ₃ (fs ₂ fz ₁)	...
				fs ₃ (fs ₂ (fs ₁ fz ₀))	...
					...

Fin zero is empty, and each $\text{Fin} (\text{suc } n)$ is made by embedding the n 'old' elements of $\text{Fin } n$, using fs_n , and adding a 'new' element fz_n . $\text{Fin } n$ provides a representation of numbers *bounded by* n , which can be used as 'array subscripts':

$$\underline{\text{let}} \left(\frac{xs : \text{Vec } n X ; i : \text{Fin } n}{\text{vproj } xs i : X} \right) ; \text{vproj } xs i \leftarrow \underline{\text{rec}} xs \{$$

$$\text{vproj } xs i \leftarrow \underline{\text{case}} xs \{$$

$$\text{vproj } \text{vnil } i \leftarrow \underline{\text{case}} i$$

$$\text{vproj } (\text{vcons } x xs) i \leftarrow \underline{\text{case}} i \{$$

$$\text{vproj } (\text{vcons } x xs) \text{fz} \Rightarrow x$$

$$\text{vproj } (\text{vcons } x xs) (\text{fs } i)$$

$$\Rightarrow \text{vproj } xs i \}}}$$

We need not fear projection from vnil , for we can dismiss $i : \text{Fin zero}$, as a harmless fiction. Of course, we could have analysed the arguments the other way

around:

$$\begin{aligned} \mathbf{vproj} \ xs \ i &\Leftarrow \underline{\mathbf{rec}} \ xs \ { \\ \mathbf{vproj} \ xs \ i &\Leftarrow \underline{\mathbf{case}} \ i \ { \\ \mathbf{vproj} \ xs \ \mathbf{fz} &\Leftarrow \underline{\mathbf{case}} \ xs \ { \\ \mathbf{vproj} \ (\mathbf{vcons} \ x \ xs) \ \mathbf{fz} &\Rightarrow x \ } \\ \mathbf{vproj} \ xs \ (\mathbf{fs} \ i) &\Leftarrow \underline{\mathbf{case}} \ xs \ { \\ \mathbf{vproj} \ (\mathbf{vcons} \ x \ xs) \ (\mathbf{fs} \ i) &\Rightarrow \mathbf{vproj} \ xs \ i \ } \}} \end{aligned}$$

Here, inspecting i forces n to be non-**zero** in each case, so xs can only be a **vcons**. The same result is achieved either way, but in both definitions, we rely on the impact the first case analysis has on the possibilities for the second. It may seem a tautology that dependent case analyses are not independent, but its impact is profound. We should certainly ask whether the traditional **case** expression, only expressing the patterns of its scrutinee, is as appropriate as it was in the past.

3.2 Refining Programming Problems

Our unification tables give some intuition to what is happening with case analysis. In Thierry Coquand’s presentation of dependent pattern matching [13], constructor case analysis is hard-wired and unification is built into the typing rules. In Epigram, we have the more generic notion of refining a programming problem by an *eliminator*, $\Leftarrow e$. If we take a closer look at the elaboration of this construct, we’ll see how unification arises and is handled *inside* the type theory. I’ll maintain both the general case and the **vtail** example side by side.

As we saw with **plus**, when we say⁸

$$\underline{\mathbf{let}} \left(\frac{\Gamma}{\mathbf{f} \ \Gamma : R} \right) \quad \Bigg| \quad \underline{\mathbf{let}} \left(\frac{ys : \mathbf{Vec} \ (\mathbf{suc} \ m) \ Y}{\mathbf{vtail} \ ys : \mathbf{Vec} \ m \ Y} \right)$$

Epigram initiates the development of a proof

$$\diamond \mathbf{f} : \forall \Gamma \Rightarrow \langle \mathbf{f} \ \Gamma : R \rangle \quad \Bigg| \quad \diamond \mathbf{vtail} : \forall m : \mathbf{Nat}; Y : \star; ys : \mathbf{Vec} \ (\mathbf{suc} \ m) \ Y \Rightarrow \langle \mathbf{vtail}_m \ Y \ ys : \mathbf{Vec} \ m \ Y \rangle$$

The general form of a subproblem in this development is

$$\diamond \mathbf{fsub} : \forall \Delta \Rightarrow \langle \mathbf{f} \ \vec{p} : T \rangle \quad \Bigg| \quad \diamond \mathbf{vtail} : \forall m : \mathbf{Nat}; Y : \star; ys : \mathbf{Vec} \ (\mathbf{suc} \ m) \ Y \Rightarrow \langle \mathbf{vtail}_m \ Y \ ys : \mathbf{Vec} \ m \ Y \rangle$$

where \vec{p} are *patterns*—expressions over the variables in Δ . In the example, I’ve chosen the initial patterns given by **vtails** formal parameters. Note that patterns

⁸ I write Greek capitals for sequences of variables with type assignments in binders and also for their unannotated counterparts as argument sequences.

in Epigram are *not* a special subclass of expression. Now let's proceed

$$\begin{array}{l|l}
\mathbf{f} \vec{p} \leftarrow e & \mathbf{vtail} \textit{ys} \leftarrow \underline{\text{case}} \textit{ys} \\
e : \forall P : \forall \Theta \Rightarrow \star & \underline{\text{case}} \textit{ys} : \forall P : \forall n; X; xs : \mathbf{Vec} \ n \ X \Rightarrow \star \\
\quad m_1 : \forall \Delta_1 \Rightarrow P \ \vec{s}_1 & \quad m_1 : \forall _X : \star \Rightarrow P \ \mathbf{zero} \ X \ \mathbf{vnil} \\
\quad \vdots & \quad m_2 : \forall _X : \star; _n : \mathbf{Nat} \\
\quad m_n : \forall \Delta_n \Rightarrow P \ \vec{s}_n & \quad \quad x : X; xs : \mathbf{Vec} \ n \ X \\
\Rightarrow P \ \vec{t} & \quad \Rightarrow P \ (\mathbf{suc} \ n) \ X \ (\mathbf{vcons} \ x \ xs) \\
& \Rightarrow P \ (\mathbf{suc} \ m) \ Y \ \textit{ys}
\end{array}$$

We call P the *motive*—it says what we gain from the elimination. In particular, we'll have a proof of P for the \vec{t} . The m_i are the *methods* by which the motive is to be achieved for each \vec{s}_i . James McKinna taught me to choose this motive:

$$\begin{array}{l|l}
\mathbf{P} \rightsquigarrow \lambda \Theta \Rightarrow & \mathbf{P} \rightsquigarrow \lambda n; X; xs \Rightarrow \\
\quad \forall \Delta \Rightarrow \Theta = \vec{t} \rightarrow & \quad \forall m : \mathbf{Nat}; Y : \star; ys : \mathbf{Vec} \ (\mathbf{suc} \ m) \ Y \\
\quad \langle \mathbf{f} \vec{p} : T \rangle & \quad \Rightarrow n = (\mathbf{suc} \ m) \rightarrow X = Y \rightarrow xs = ys \rightarrow \\
& \quad \langle \mathbf{vtail}_{m \ Y} \ \textit{ys} : \mathbf{Vec} \ m \ Y \rangle
\end{array}$$

This is just Henry Ford's old joke. Our motive is to produce a proof of $\forall \Delta \Rightarrow \langle \mathbf{f} \vec{p} : T \rangle$, for 'any Θ we like as long as it's \vec{t} '—the \vec{t} are the only Θ we keep in stock. For our example, that means 'any vector you like as long as it's nonempty and its elements are from Y '. This $=$ is **heterogeneous equality**,⁹ which allows any elements of arbitrary types to be proclaimed equal. Its one constructor, **refl**, says that a thing is equal to itself.

$$\frac{s : S; t : T}{s = t : \star} \quad \frac{}{\mathbf{refl} : t = t}$$

Above, the types of xs and ys are different, but they will unify if we can solve the prior equations. Hypothetical equations don't change the internal rules by which the typechecker compares types—this is lucky, as hypotheses can lie.

If we can construct the methods, \mathbf{m}_i , then we're done:

$$\begin{array}{l|l}
\diamond \mathbf{fsub} \rightsquigarrow \lambda \Delta \Rightarrow e \ \mathbf{P} \ \mathbf{m}_1 \ \dots \ \mathbf{m}_n & \diamond \mathbf{vtail} \rightsquigarrow \lambda m; Y; ys \Rightarrow (\underline{\text{case}} \ \textit{ys}) \ \mathbf{P} \ \mathbf{m}_1 \ \mathbf{m}_2 \\
\quad \Delta \ \mathbf{refl} \ \dots \ \mathbf{refl} & \quad m \ Y \ \textit{ys} \ \mathbf{refl} \ \mathbf{refl} \ \mathbf{refl} \\
\quad : \forall \Delta \Rightarrow \langle \mathbf{f} \vec{p} : T \rangle & \quad : \forall m : \mathbf{Nat}; Y : \star; ys : \mathbf{Vec} \ (\mathbf{suc} \ m) \ Y \\
& \quad \Rightarrow \langle \mathbf{vtail}_{m \ Y} \ \textit{ys} : \mathbf{Vec} \ m \ Y \rangle
\end{array}$$

But what are the methods? We must find, for each i

$$\mathbf{m}_i : \forall \Delta_i; \Delta \Rightarrow \vec{s}_i = \vec{t} \rightarrow \langle \mathbf{f} \vec{p} : T \rangle$$

⁹ also known as 'John Major' equality [23]

In our example, we need

$$\begin{aligned}
\mathbf{m}_1 & : \forall _X; _m; _Y; ys : \mathbf{Vec} (\mathbf{suc} \ m) \ Y \Rightarrow \\
& \quad \mathbf{zero} = (\mathbf{suc} \ m) \rightarrow X = Y \rightarrow \mathbf{vnil} = ys \rightarrow \\
& \quad \langle \mathbf{vtail}_{m \ Y} \ ys : \mathbf{Vec} \ m \ Y \rangle \\
\mathbf{m}_2 & : \forall _X; _n; x; xs : \mathbf{Vec} \ n \ X; _m; _Y; xs : \mathbf{Vec} (\mathbf{suc} \ m) \ Y \Rightarrow \\
& \quad (\mathbf{suc} \ n) = (\mathbf{suc} \ m) \rightarrow X = Y \rightarrow (\mathbf{vcons} \ x \ xs) = ys \rightarrow \\
& \quad \langle \mathbf{vtail}_{m \ Y} \ ys : \mathbf{Vec} \ m \ Y \rangle
\end{aligned}$$

Look at the equations! They express exactly the unification problems for case analysis which we tabulated informally. Now to solve them: the rules of first-order unification for data constructors—see figure 1—are derivable in UTT. Each rule

deletion	$P \rightarrow$ $x = x \rightarrow P$
conflict	$\mathbf{chalk} \ \vec{s} = \mathbf{cheese} \ \vec{t} \rightarrow P$
injectivity	$(\vec{s} = \vec{t} \rightarrow P) \rightarrow$ $\mathbf{chalk} \ \vec{s} = \mathbf{chalk} \ \vec{t} \rightarrow P$
substitution	$P \ t \rightarrow$ $x, t : T; x \notin FV(t)$ $x = t \rightarrow P \ x$
cycle	$x = t \rightarrow P$ x constructor-guarded in t

Fig. 1. derivable unification rule schemes

(read backwards) simplifies a problem with an equational hypothesis. We apply these simplifications to the method types. The **conflict** and **cycle** rules dispose of ‘impossible case’ subproblems. Meanwhile, the **substitution** rule instantiates pattern variables. In general, the equations $\vec{s}_i = \vec{t}$ will be reduced as far as possible by first-order unification, and either the subproblem will be dismissed, or it will yield some substitution, instantiating the patterns \vec{p} .

In our example, the **vnil** case goes by **conflict**, and the **vcons** case becomes:

$$\forall _Y; _m; x; Y; xs : \mathbf{Vec} \ Y \ m \Rightarrow \langle \mathbf{vtail}_{Y \ m} (\mathbf{vcons} \ x \ xs) : \mathbf{Vec} \ Y \ m \rangle$$

After ‘cosmetic renaming’ gives x and xs names more like the original ys , we get

$$\mathbf{vtail} (\mathbf{vcons} \ y \ ys) \ \boxed{[]}$$

To summarize, elaboration of $\Leftarrow e$ proceeds as follows:

- (1) choose a motive with equational constraints;
- (2) simplify the constraints in the methods by first-order unification;
- (3) leave the residual methods as the subproblems to be solved by subprograms.

In the presence of defined functions and higher types, unification problems won't always be susceptible to first-order unification, but Epigram will make what progress it can and leave the remaining equations unsolved in the hypotheses of subproblems—later analyses may reduce them to a soluble form. Moreover, there is no reason in principle why we should not consider a constraint-solving procedure which can be customized by user-supplied rules.

3.3 Reflection on Inspection

We're not used to thinking about what functions really tell us, because simple types don't say much about values, statically. For example, we could write

$$\underline{\text{let}} \left(\frac{n : \text{Nat}}{\text{nonzero } n : \text{Bool}} \right) ; \text{nonzero } n \Leftarrow \text{case } n \{ \begin{array}{l} \text{nonzero zero} \Rightarrow \text{false} \\ \text{nonzero (suc } n) \Rightarrow \text{true} \end{array} \}$$

but suppose we have $xs : \text{Vec } n \ X$ —what do we learn by testing $\text{nonzero } n$? All we get is a **Bool**, with no direct implications for our understanding of n or xs . We are in no better position to apply vtail to xs after inspecting this **Bool** than before. Instead, if we do case analysis on n , we learn what n is *statically* as well as dynamically, and in the **suc** case we can apply vtail to xs .

Of course, we could think of writing a preVtail function which operates on any vector but requires a precondition, like

$$\underline{\text{let}} \left(\frac{xs : \text{Vec } n \ X ; q : \text{nonzero } n = \text{true}}{\text{preVtail } xs \ q : \text{Vec } [] \ X} \right)$$

I'm not quite sure what to write as the length of the returned vector, so I've left a shed: perhaps it needs some kind of predecessor function with a precondition. If we have $ys : \text{Vec } (\text{suc } m) \ Y$, then $\text{preVtail } ys \ \text{refl}$ will be well typed. We could even use this function with a more informative conditional expression:

$$\text{condInfo} : \forall P : \star ; b : \text{Bool} \Rightarrow (b = \text{true} \rightarrow P) \rightarrow (b = \text{false} \rightarrow P) \rightarrow P$$

However, this way of working is clearly troublesome.

Moreover, given a nonempty vector xs , there is more than just a stylistic difference between decomposing it with vhead and vtail and decomposing it with $(\text{case } xs)$ —the destructor functions give us an element and a shorter vector; the case analysis tells us that xs is the vcons of them, and if any types depend on xs , that might just be important. Again, we can construct a proof of $xs = \text{vcons } (\text{vhead } xs) (\text{vtail } xs)$, but this is much harder to work with.

In the main, selectors-and-destructors are poor tools for working with data on which types depend. We really need forms of inspection which yield static information. This is a new issue, so there's no good reason to believe that the old design choices remain appropriate. We need to think carefully about how to reflect data's new rôle as *evidence*.

3.4 Vectorized Applicative Programming

Now that we've seen how dependent case analysis is elaborated, let's do some more work with it. The next example shows a key difference between Epigram's implicit syntax and *parametric* polymorphism. The operation

$$\underline{\text{let}} \left(\frac{x : X}{\mathbf{vec} \ x : \mathbf{Vec} \ n \ X} \right)$$

makes a vector of copies of its argument. For any given *usage* of **vec**, the intended type determined the length, but how are we to *define* **vec**? We shall need to work by recursion on the intended length, hence we shall need to make this explicit at definition time. The following declaration achieves this:

$$\underline{\text{let}} \left(\frac{n : \mathbf{Nat} ; x : X}{\mathbf{vec}_n \ x : \mathbf{Vec} \ n \ X} \right) ; \begin{array}{l} \mathbf{vec}_n \ x \Leftarrow \underline{\text{rec}} \ n \ { \\ \mathbf{vec}_n \ x \Leftarrow \underline{\text{case}} \ n \ { \\ \mathbf{vec}_{\text{zero}} \ x \Rightarrow \mathbf{vnil} \\ \mathbf{vec}_{(\text{suc } n)} \ x \Rightarrow \mathbf{vcons} \ x \ (\mathbf{vec}_n \ x) \}} \end{array}$$

Note that in **vec**'s type signature, I explicitly declare n first, thus making it the first implicit argument: otherwise, X might happen to come first. By the way, we don't have to override the argument in the recursive call **vec** _{n} x —it's got to be a **Vec** n X —but it would perhaps be a little disconcerting to omit the n , especially as it's the key to **vec**'s structural recursion.

The following operation—vectorized application—turns out to be quite handy.

$$\underline{\text{let}} \left(\frac{fs : \mathbf{Vec} \ n \ (S \rightarrow T) ; ss : \mathbf{Vec} \ n \ S}{\mathbf{va} \ fs \ ss : \mathbf{Vec} \ n \ T} \right)$$

$$\begin{array}{l} \mathbf{va} \ fs \ ss \Leftarrow \underline{\text{rec}} \ fs \ { \\ \mathbf{va} \ fs \ ss \Leftarrow \underline{\text{case}} \ fs \ { \\ \mathbf{va} \ \mathbf{vnil} \ ss \Leftarrow \underline{\text{case}} \ ss \ { \\ \mathbf{va} \ \mathbf{vnil} \ \mathbf{vnil} \Rightarrow \mathbf{vnil} \} \\ \mathbf{va} \ (\mathbf{vcons} \ f \ fs) \ ss \Leftarrow \underline{\text{case}} \ ss \ { \\ \mathbf{va} \ (\mathbf{vcons} \ f \ fs) \ (\mathbf{vcons} \ s \ ss) \Rightarrow \mathbf{vcons} \ (f \ s) \ (\mathbf{va} \ fs \ ss) \}} \end{array}$$

As it happens, the combination of **vec** and **va** equip us with 'vectorized applicative programming', with **vec** embedding the constants, and **va** providing application. Transposition is my favourite example of this:

$$\underline{\text{let}} \left(\frac{x_{ij} : \mathbf{Vec} \ i \ (\mathbf{Vec} \ j \ X)}{\mathbf{transpose} \ x_{ij} : \mathbf{Vec} \ j \ (\mathbf{Vec} \ i \ X)} \right)$$

$$\begin{array}{l} \mathbf{transpose} \ x_{ij} \Leftarrow \underline{\text{rec}} \ x_{ij} \ { \\ \mathbf{transpose} \ x_{ij} \Leftarrow \underline{\text{case}} \ x_{ij} \ { \\ \mathbf{transpose} \ \mathbf{vnil} \Rightarrow \mathbf{vec} \ \mathbf{vnil} \\ \mathbf{transpose} \ (\mathbf{vcons} \ x_j \ x_{ij}) \Rightarrow \mathbf{va} \ (\mathbf{va} \ (\mathbf{vec} \ \mathbf{vcons}) \ x_j) \ (\mathbf{transpose} \ x_{ij}) \}} \end{array}$$

3.5 Exercises: Matrix Manipulation

Exercise 7 (vmap, vZipWith) Show how **vec** and **va** can be used to generate the vector analogues of Haskell's `map`, `zipWith`, and the rest of the family. (A glance at [16] may help.)

Exercise 8 (vdot) Implement **vdot**, the scalar product of two vectors of **Nat**s.

Now, how about matrices? Recall the vector-of-columns representation:

$$\underline{\text{let}} \left(\frac{\text{rows, cols} : \mathbf{Nat}}{\mathbf{Matrix} \text{ rows cols} : \star} \right) \mathbf{Matrix} \text{ rows cols} \Rightarrow \mathbf{Vec} \text{ cols} (\mathbf{Vec} \text{ rows Nat})$$

Exercise 9 (zero, identity) How would you compute the zero matrix of a given size? Also implement a function to compute any identity matrix.

Exercise 10 (matrix by vector) Implement matrix-times-vector multiplication. (ie, interpret a **Matrix** $m \ n$ as a linear map $\mathbf{Vec} \ n \ \mathbf{Nat} \rightarrow \mathbf{Vec} \ m \ \mathbf{Nat}$.)

Exercise 11 (matrix by matrix) Implement matrix-times-matrix multiplication. (ie, implement composition of linear maps.)

Exercise 12 (monad) (Mainly for Haskellers.) It turns out that for each n , $\mathbf{Vec} \ n$ is a monad, with **vec** playing the part of `return`. What should the corresponding notion of `join` do? What plays the part of `ap`?

3.6 Exercises: Finite Sets

Exercise 13 (fmax, fweak) Implement **fmax** (each nonempty set's maximum value) and **fweak** (the function preserving **fz** and **fs**, incrementing the index).

$$\underline{\text{let}} \left(\frac{}{\mathbf{fmax}_n : \mathbf{Fin} (\mathbf{suc} \ n)} \right) \quad \Bigg| \quad \underline{\text{let}} \left(\frac{i : \mathbf{Fin} \ n}{\mathbf{fweak} \ i : \mathbf{Fin} (\mathbf{suc} \ n)} \right)$$

You should find that **fmax** and **fweak** partition the finite sets, just as **fz** and **fs** do. Imagine how we might pretend they're an alternative set of constructors...

Exercise 14 (vtab) Implement **vtab**, the inverse of **vproj**, tabulating a function over finite sets as a vector.

$$\underline{\text{let}} \left(\frac{n : \mathbf{Nat} ; f : \mathbf{Fin} \ n \rightarrow X}{\mathbf{vtab}_n f : \mathbf{Vec} \ n \ X} \right)$$

Note that **vtab** and **vproj** offer alternative definitions of matrix operations.

Exercise 15 (OPF, opf) Devise an inductive family, **OPF** $m \ n$ which gives a unique first-order representation of exactly the order-preserving functions in $\mathbf{Fin} \ m \rightarrow \mathbf{Fin} \ n$. Give your family a semantics by implementing

$$\underline{\text{let}} \left(\frac{f : \mathbf{OPF} \ m \ n ; i : \mathbf{Fin} \ m}{\mathbf{opf} \ f \ i : \mathbf{Fin} \ n} \right)$$

Exercise 16 (iOPF, cOPF) *Implement identity and composition:*

$$\underline{\text{let}} \left(\frac{}{\mathbf{iOPF}_n : \mathbf{OPF} \ n \ n} \right) \quad \Bigg| \quad \underline{\text{let}} \left(\frac{f \ \mathbf{OPF} \ m \ n ; g \ \mathbf{OPF} \ l \ m}{\mathbf{cOPF} \ f \ g : \mathbf{OPF} \ l \ n} \right)$$

Which laws should relate **iOPF**, **cOPF** and **opf**?

4 Representing Syntax

The **Fin** family can represent de Bruijn indices in nameless expressions [14]. As Françoise Bellegarde and James Hook observed in [6], and Richard Bird and Ross Paterson were able to implement in [9], you can do this in Haskell, up to a point—here are the λ -terms with *free* variables given by **v**:

```
data Term v = Var v
            | App (Term v) (Term v)
            | Lda (Term (Maybe v))
```

Under a **Lda**, we use **(Maybe v)** as the variable set for the body, with **Nothing** being the new free variable and **Just** embedding the old free variables. Renaming is just **fmap**, and substitution is just the monadic ‘bind’ operator **>>=**.

However, **Term** is a bit *too* polymorphic. We can’t see the *finiteness* of the variable context over which a term is constructed. In Epigram, we can take the number of free variables to be a number **n**, and choose variables from **Fin n**.

$$\underline{\text{data}} \left(\frac{n : \mathbf{Nat}}{\mathbf{Tm} \ n : \star} \right)$$

$$\underline{\text{where}} \left(\frac{i : \mathbf{Fin} \ n}{\mathbf{var} \ i : \mathbf{Tm} \ n} \right) ; \left(\frac{f, s : \mathbf{Tm} \ n}{\mathbf{app} \ f \ s : \mathbf{Tm} \ n} \right) ; \left(\frac{t : \mathbf{Tm} \ (\mathbf{suc} \ n)}{\mathbf{lda} \ t : \mathbf{Tm} \ n} \right)$$

The **n** in **Term n** indicates the number of variables available for term formation: we can explain how to λ -lift a term, by abstracting over all the available variables:

$$\underline{\text{let}} \left(\frac{t : \mathbf{Tm} \ n}{\mathbf{ldaLift}_n \ t : \mathbf{Tm} \ \mathbf{zero}} \right) ; \mathbf{ldaLift}_n \ t \Leftarrow \underline{\text{rec}} \ n \ {$$

$$\mathbf{ldaLift}_n \ t \Leftarrow \underline{\text{case}} \ n \ {$$

$$\mathbf{ldaLift}_{\mathbf{zero}} \ t \Rightarrow t$$

$$\mathbf{ldaLift}_{(\mathbf{suc} \ n)} \ t \Rightarrow \mathbf{ldaLift}_n \ (\mathbf{lda} \ t) \ } \}$$

Not so long ago, we were quite excited about the power of non-uniform datatypes to capture useful structural invariants. Scoped de Bruijn terms gave a good example, but most of the others proved more awkward even than the ‘fake’ dependent types you can cook up using type classes [24].

Real dependent types achieve more with less fuss. This is mainly due to the flexibility of inductive families. For example, if you wanted to add ‘weakening’ to delay explicitly the shifting of a term as you push it under a binder—in Epigram, but not Haskell or Cayenne, you could add the constructor

$$\left(\frac{t : \mathbf{Tm} \ n}{\mathbf{weak} \ t : \mathbf{Tm} \ (\mathbf{suc} \ n)} \right)$$

4.1 Exercises: Renaming and Substitution

If $\text{Fin } m$ is a variable set, then some $\rho : \text{Fin } m \rightarrow \text{Fin } n$ is a *renaming*. If we want to apply a renaming to a term, we need to be able to push it under a **lda**. Hence we need to *weaken* the renaming, mapping the new source variable to the new target variable, and renaming as before on the old variables.¹⁰

$$\underline{\text{let}} \left(\frac{\rho : \text{Fin } m \rightarrow \text{Fin } n ; i : \text{Fin } (\text{suc } m)}{\mathbf{wren } \rho i : \text{Fin } (\text{suc } n)} \right) \quad \mathbf{wren } \rho i \Leftarrow \underline{\text{case}} i \{$$

$$\mathbf{wren } \rho \mathbf{fz} \Rightarrow \mathbf{fz}$$

$$\mathbf{wren } \rho (\mathbf{fs } i) \Rightarrow \mathbf{fs } (\rho i) \}$$

You get to finish the development.

Exercise 17 (ren) Use **wren** to help you implement the renaming traversal

$$\underline{\text{let}} \left(\frac{\rho : \text{Fin } m \rightarrow \text{Fin } n ; t : \text{Tm } m}{\mathbf{ren } \rho t : \text{Tm } n} \right)$$

Now repeat the pattern for substitutions—functions from variables to terms.

Exercise 18 (wsub, sub) Develop weakening for substitutions, then use it to go under **lda** in the traversal:

$$\underline{\text{let}} \left(\frac{\sigma : \text{Fin } m \rightarrow \text{Tm } n ; i : \text{Fin } (\text{suc } m)}{\mathbf{wsub } \sigma i : \text{Tm } (\text{suc } n)} \right)$$

$$\underline{\text{let}} \left(\frac{\sigma : \text{Fin } m \rightarrow \text{Tm } n ; t : \text{Tm } m}{\mathbf{sub } \sigma t : \text{Tm } n} \right)$$

Exercise 19 (For the brave.) Refactor this development, abstracting the weakening-then-traversal pattern. If you need a hint, see chapter 7 of [23].

4.2 Stop the World I Want to Get Off! (a first try at typed syntax)

We've seen untyped λ -calculus: let's look at how to enforce stronger invariants, by representing a *typed* λ -calculus. Recall the rules of the simply typed λ -calculus:

$$\frac{}{\Gamma ; x \in \sigma ; \Gamma' \vdash x \in \sigma} \quad \frac{\Gamma ; x \in \sigma \vdash t \in \tau}{\Gamma \vdash \lambda x \in \sigma . t \in \sigma \supset \tau} \quad \frac{\Gamma \vdash f \in \sigma \supset \tau \quad \Gamma \vdash s \in \sigma}{\Gamma \vdash f s \in \tau}$$

Well-typed terms are defined with respect to a context and a type. Let's just turn the rules into data! I add a base type, to make things more concrete.

$$\underline{\text{data}} \left(\frac{}{\text{SType} : \star} \right) \quad \underline{\text{where}} \left(\frac{}{\mathbf{sNat} : \text{SType}} \right) ; \left(\frac{\sigma, \tau : \text{SType}}{\mathbf{sFun } \sigma \tau : \text{SType}} \right)$$

¹⁰ Categorists! Note **wren** makes **suc** a functor in the category of **Fin**-functions. What other structure can you sniff out here?

We could use `Vec` for contexts, but I prefer contexts which grow on the right.

$$\underline{\text{data}} \left(\frac{n : \text{Nat}}{\text{SCtxt } n : \star} \right) \underline{\text{where}} \left(\frac{\text{empty} : \text{SCtxt zero}}{\left(\frac{\Gamma : \text{SCtxt } n ; \sigma : \text{SType}}{\text{bind } \Gamma \sigma : \text{SCtxt (suc } n)} \right)} \right)$$

Now, assuming we have a projection function `sproj`, defined in terms of `SCtxt` and `Fin` the way we defined `vproj`, we can just turn the inference rules of the typing relation into constructors:

$$\underline{\text{data}} \left(\frac{\Gamma : \text{SCtxt } n ; \tau : \text{SType}}{\text{STm } \Gamma \tau : \star} \right) \\ \underline{\text{where}} \left(\frac{i : \text{Fin } n}{\text{svar } i : \text{STm } \Gamma (\text{sproj } \Gamma i)} \right) ; \left(\frac{t : \text{STm (bind } \Gamma \sigma) \tau}{\text{slda } t : \text{STm } \Gamma (\text{sFun } \sigma \tau)} \right) \\ \left(\frac{f : \text{STm } \Gamma (\text{sFun } \sigma \tau) ; s : \text{STm } \Gamma \sigma}{\text{sapp } f s : \text{STm } \Gamma \tau} \right)$$

This is a precise definition of the simply-typed λ -terms. But is it any good? Well, just try writing programs with it.

How would you implement *renaming*? As before, we could represent a renaming as a function $\rho : \text{Fin } m \rightarrow \text{Fin } n$. Can we rename a term in `STm Γ τ` to get a `STm Δ τ` , where $\Gamma : \text{SCtxt } m$ and $\Delta : \text{SCtxt } n$? Here comes the crunch:

$$\dots \text{ren}_{n \Gamma \Delta} \rho (\text{svar } i) \Rightarrow \text{svar } (\rho i)$$

The problem is that `svar i : STm Γ (sproj Γ i)`, so we want a `STm Δ (sproj Γ i)` on the right, but we've got a `STm Δ (sproj Δ (ρ i))`. We need to know that ρ is type-preserving! Our choice of variable representation prevents us from building this into the type of ρ . We are forced to state an extra condition:

$$\forall i : \text{Fin } m \Rightarrow \text{sproj } \Gamma i = \text{sproj } \Delta (\rho i)$$

We'll need to repair our program by rewriting with this proof.¹¹ But it's worse than that! When we move under a `slda`, we'll lift the renaming, so we'll need a *different* property:

$$\forall i' : \text{Fin (suc } m) \Rightarrow \text{sproj (bind } \Gamma \sigma) i' = \text{sproj (bind } \Delta \sigma) (\text{lift } \rho i')$$

This follows from the previous property, but it takes a little effort. My program has just filled up with ghastly theorem-proving. Don't dependent types make life a nightmare? Stop the world I want to get off!

If you're not afraid of hard work, you can carry on and make this program work. I think discretion is the better part of valour—let's solve the problem instead. We're working with *typed* terms but *untyped* variables, and our function which gives types to variables does not connect the variable clearly to the context. For all we know, `sproj` always returns `sNat`! No wonder we need 'logical superstructure' to recover the information we've thrown away.

¹¹ I shan't show how to do this, as we shall shortly avoid the problem.

4.3 Dependent Types to the Rescue

Instead of using a program to assign types to variables and then reasoning about it, let's just have *typed* variables, as Thorsten Altenkirch and Bernhard Reus [2].

$$\begin{array}{c} \text{data} \left(\frac{\Gamma : \text{SCtxt } n; \tau : \text{SType}}{\text{SVar } \Gamma \tau : \star} \right) \\ \text{where} \left(\frac{}{\text{vz} : \text{SVar}(\text{bind } \Gamma \sigma) \sigma} \right) ; \left(\frac{i : \text{SVar } \Gamma \tau}{\text{vs } i : \text{SVar}(\text{bind } \Gamma \sigma) \tau} \right) \end{array}$$

This family strongly resembles `Fin`. Its constructors target only nonempty contexts; it has one constructor which references the ‘newest’ variable; the other constructor embeds the ‘older’ variables. You may also recognize this family as an inductive definition of context *membership*. Being a variable *means* being a member of the context. `Fin` just gives a data representation for variables without their meaning. Now we can replace our awkward `svar` with

$$\left(\frac{i : \text{SVar } \Gamma \tau}{\text{svar } i : \text{STm } \Gamma \tau} \right)$$

A renaming *from* Γ *to* Δ becomes an element of

$$\forall \tau : \text{SType} \Rightarrow \text{SVar } \Gamma \tau \rightarrow \text{SVar } \Delta \tau$$

Bad design makes for hard work, whether you're making can-openers, doing mathematics or writing programs. It's often tempting to imagine that once we've made our representation of data tight enough to rule out meaningless values, our job is done and things should just work out. This experience teaches us that more is required—we should use types to give meaningful values their meaning. `Fin` contains the right data, but `SVar` actually explains it.

Exercise 20 *Construct simultaneous renaming and simultaneous substitution for this revised definition of `STm`. Just lift the pattern from the untyped version!*

4.4 Is Looking Seeing?

It's one thing to define data structures which *enforce* invariants and to write programs which *respect* invariants, but how can we *establish* invariants?

We've seen how to use a finite set to index a vector, enforcing the appropriate bounds, but what if we only have a *number*, sent to us from the outside world? We've seen how to write down the `STms`, but what if we've read in a program from a file? How do we compute its type-safe representation if it has one?

If we want to index a `Vec n X` by $m : \text{Nat}$, it's no good testing the Boolean $m < n$. The value `true` or `false` won't explain whether m can be represented by some $i : \text{Fin } n$. If we have a $f : \text{STm } \Gamma (\text{sFun } \sigma \tau)$ and some $a : \text{STm } \Gamma \alpha$, we could check Boolean equality $\sigma == \alpha$, but `true` doesn't make α into σ , so we can't construct `sapp f a`.

Similar issues show up in the ‘Scrap Your Boilerplate’ library of dynamically typed traversal operators by Ralf Lämmel and Simon Peyton Jones [18]. The whole thing rests on a ‘type safe cast’ operator, comparing types at run time:

```

cast :: (Typeable a, Typeable b) => a -> Maybe b
cast x = r
  where
    r = if typeOf x == typeOf (get r)
        then Just (unsafeCoerce x)
        else Nothing

get :: Maybe a -> a
get x = undefined

```

This program does not, of itself, make sense. The best we can say is that *we* can make sense of it, provided `typeOf` has been correctly implemented. The machine *looks* at the types but does not *see* when they are the same, hence the `unsafeCoerce`. The significance of the test is obscure, so blind obedience is necessary. Of course, *I* trust them, but I think they could aspire for better.

The trouble is that representing the result of a computation is not enough: you need to know the *meaning* of the computation if you want to justify its consequences. A Boolean is a bit uninformative. To see when we look, we need a new way of looking. Take the vector indexing example. We can explain which number is represented by a given $i : \text{Fin } n$ by *forgetting* its bound:

$$\text{let } \left(\frac{i : \text{Fin } n}{\text{fFin } i : \text{Nat}} \right) ; \text{ fFin } i \leftarrow \text{rec } i \{ \begin{array}{l} \text{fFin } i \leftarrow \text{case } i \{ \\ \text{fFin } fz \Rightarrow \text{zero} \\ \text{fFin } (fs \ i) \Rightarrow \text{suc } (\text{fFin } i) \} \end{array} \right.$$

Now, for a given n and m , m is either

- $(\text{fFin } i)$ for some $i : \text{Fin } n$, or
- $(\text{plus } n \ m')$ for some $m' : \text{Nat}$

Our types can talk about values—we can *say* that!

$$\text{checkBound } n \ m : \forall P : \text{Nat} \rightarrow \text{Nat} \rightarrow \star \Rightarrow \begin{array}{l} (\forall n : \text{Nat}; i : \text{Fin } n \Rightarrow P \ n \ (\text{fFin } i)) \rightarrow \\ (\forall n, m' : \text{Nat} \Rightarrow P \ n \ (\text{plus } n \ m')) \rightarrow \\ P \ n \ m \end{array}$$

That’s to say: ‘whatever P you want to do with n and m , it’s enough to explain P for n and $(\text{fFin } i)$ and also for n and $(\text{plus } n \ m')$. Or ‘you can match n, m against the *patterns*, $n, (\text{fFin } i)$ and $n, (\text{plus } n \ m')$. I designed the above type to look like a `case` principle, so that I can program with it. Note that I don’t just get an element either of $(\text{Fin } i)$ or of Nat from an anonymous informant; it

really is my very own n and m which get analysed—the type says so! If I have **checkBound**, then I can check m like this:

$$\begin{aligned} & \underline{\text{let}} \left(\frac{xs : \text{Vec } n \ X ; m : \text{Nat}}{\text{mayProj}_n \ xs \ m : \text{Maybe } X} \right) \\ & \text{mayProj}_n \ xs \ m \Leftarrow \text{checkBound } n \ m \ \{ \\ & \quad \text{mayProj}_n \ xs \ (\text{fFin } i) \Rightarrow \text{just } (\text{vproj } xs \ i) \\ & \quad \text{mayProj}_n \ xs \ (\text{plus } n \ m') \Rightarrow \text{nothing} \} \end{aligned}$$

In one case, we get a bounded i , so we can apply bounds-safe projection. In the other, we clearly fail. Moreover, if the return type were to depend on m , that’s fine: not only do *we* see what m must be, Epigram sees it too! But **checkBound** has quite a complicated higher-order type. Do I really expect you to dump good old $m < n$ for some bizarre functional? Of course I don’t: I’ll now explain the straightforward first-order way to construct **checkBound**.

4.5 A Funny Way of Seeing Things

Constructor case analysis is the normal way of seeing things. Suppose I have a funny way of seeing things. We know that \Leftarrow doesn’t care—a ‘way of seeing things’ is expressed by a type and interpreted as a way of decomposing a programming problem into zero or more subproblems. But how do I establish that my funny way of seeing at things makes sense?

Given $n, m : \text{Nat}$, we want to see m as either $(\text{fFin } i)$ for some $i : \text{Fin } n$, or else some $(\text{plus } n \ m')$. We can write a predicate which characterizes the n and m for which this is possible—it’s possible for the very patterns we want.¹²

$$\begin{aligned} & \underline{\text{data}} \left(\frac{n, m : \text{Nat}}{\text{BoundCheck } n \ m : \star} \right) \\ & \underline{\text{where}} \left(\frac{i : \text{Fin } n}{\text{inBound } i : \text{BoundCheck } n \ (\text{fFin } i)} \right) \\ & \left(\frac{m' : \text{Nat}}{\text{outOfBound } m' : \text{BoundCheck } n \ (\text{plus } n \ m')} \right) \end{aligned}$$

A value $bc : \text{BoundCheck } n \ m$ tells us something about n and m , and it’s just n and m that we care about here— bc is just a means to this end. The eliminator ($\underline{\text{case}} \ bc$) expects a motive abstracting over n , m and bc , allowing us to inspect bc also. If we restrict the motive to see only n and m , we get

$$\begin{aligned} & \lambda P : \text{Nat} \rightarrow \text{Nat} \rightarrow \star \Rightarrow (\underline{\text{case}} \ bc) (\lambda n' ; m' ; bc' \Rightarrow P \ n' \ m') \\ & : \forall P : \text{Nat} \rightarrow \text{Nat} \rightarrow \star \Rightarrow \\ & \quad (\forall n : \text{Nat} ; i : \text{Fin } n \Rightarrow P \ n \ (\text{fFin } i)) \rightarrow \\ & \quad (\forall n, m' : \text{Nat} \Rightarrow P \ n \ (\text{plus } n \ m')) \rightarrow \\ & \quad P \ n \ m \end{aligned}$$

¹² I forgot that I’m *programming*: of course, I mean ‘datatype family’.

and that’s exactly the type of `checkBound` $n m$. This construction on a predicate is sufficiently useful that Epigram gives it a special name, (`view bc`). That’s to say, the machine-generated eliminator which just looks at `BoundCheck`’s *indices* in terms of its constructors. Logically, `view` gives a datatype family its **relation induction** principle. But to use this ‘view’, we need $bc : \text{BoundCheck } n m$. That is, we must show that every n and m are checkable in this way:

$$\text{let } \left(\frac{}{\text{boundCheck } n m : \text{BoundCheck } n m} \right)$$

```

boundCheck n m ← rec n {
  boundCheck n m ← case n {
    boundCheck zero m ⇒ outOfBound m
    boundCheck (suc n) m ← case m {
      boundCheck (suc n) zero ⇒ inBound fz
      boundCheck (suc n) (suc m) ← view (boundCheck n m) {
        boundCheck (suc n) (suc (fFin i)) ⇒ inBound (fs i)
        boundCheck (suc n) (suc (plus n m')) ⇒ outOfBound m' }}}}

```

There’s no trouble using the view we’re trying to establish: the recursive call is structural, but used in an eliminator rather than a return value. This function works much the way *subtraction* works. The only difference is that it has a type which establishes a connection between the output to the function and its inputs, shown directly in the patterns! We may now take

$$\text{checkBound } n m \rightsquigarrow \text{view } (\text{boundCheck } n m)$$

4.6 Patterns Forget; Matching Is Remembering

What has ‘pattern matching’ become? In general, a pattern is a *forgetful* operation. Constructors like `zero` and `suc` forget *themselves*—you can’t tell from the type `Nat`, which constructor you’ve got. Case analysis remembers what constructors forget. And so it is with our funny patterns: the `fFin` function forgets bounded whilst (`plus n m'`) forgets by how much its output exceeds n . Our view remembers what these patterns forget.

The difference between Epigram views and Phil Wadler’s views [38] is that Epigram views cannot lie. Epigram views talk directly about the values being inspected in terms of the forgetful operations which generate them. Wadler’s views ascribe that informative significance to an independent value, whether or not it’s justified. We shouldn’t criticize Wadler for this—dependent types can see where simple types can only look. Of course, to work with dependent types, we *need* to be able to see. If we want to generate values in types which enforce strong invariants, we need to see that those invariants hold.

Exercise 21 Show that `fmax` and `fweak` cover `Fin` by constructing a view.

5 Well Typed Programs which Don't Go Wrong

Let's have a larger example of derivable pattern matching—building simply-typed terms in the **STm** family by typechecking 'raw' untyped terms from

$$\begin{array}{c} \text{data} \left(\frac{n : \text{Nat}}{\text{RTm } n : \star} \right) \\ \text{where} \left(\frac{i : \text{Fin } n}{\text{rvar } i : \text{RTm } n} \right) ; \left(\frac{f, s : \text{RTm } n}{\text{rapp } f s : \text{RTm } n} \right) \\ \left(\frac{\sigma : \text{SType} ; b : \text{RTm } (\text{suc } n)}{\text{rlda } \sigma b : \text{RTm } n} \right) \end{array}$$

Typechecking is a form of looking. It relies on two auxiliary forms of looking—looking up a variable in the context, and checking that two types are the same. Our **svar** constructor takes context-references expressed in terms of **SVar**, and our **sapp** constructor really needs the domain of the function to be the same as the type of the argument, so just looking is not enough. Let's see.

An **SVar** is a context-reference; a **Fin** is merely a context-pointer. We can clearly turn a reference into a pointer by forgetting what's referred to:

$$\text{let} \left(\frac{\Gamma : \text{SCtxt } n ; i : \text{SVar } \Gamma \tau}{\text{fV } \tau i : \text{Fin } n} \right) ; \begin{array}{l} \text{fV } \tau i \Leftarrow \text{rec } i \{ \\ \text{fV } \tau i \Leftarrow \text{case } i \{ \\ \text{fV } \tau \text{vz} \Rightarrow \text{fz} \\ \text{fV } \tau (\text{vs } i) \Rightarrow \text{fs } (\text{fV } \tau i) \} \} \end{array}$$

Why is τ an explicit argument? Well, the point of writing this forgetful map is to define a notion of *pattern* for finite sets which characterizes projection. We need to see the information which the pattern throws away. Let's establish the view—it's just a more informative **vproj**, telling us not only the projected thing, but that it is indeed the projection we wanted.

$$\begin{array}{c} \text{data} \left(\frac{\Gamma : \text{SCtxt } n ; i : \text{Fin } n}{\text{Find } \Gamma i : \star} \right) \text{ where} \left(\frac{i : \text{SVar } \Gamma \tau}{\text{found } \tau i : \text{Find } \Gamma (\text{fV } \tau i)} \right) \\ \\ \text{let} \left(\frac{\text{find } \Gamma i : \text{Find } \Gamma i}{\text{find } \Gamma i \Leftarrow \text{rec } \Gamma \{ \\ \text{find } \Gamma i \Leftarrow \text{case } i \{ \\ \text{find } \Gamma \text{fz} \Leftarrow \text{case } \Gamma \{ \\ \text{find } (\text{bind } \Gamma \sigma) \text{fz} \Rightarrow \text{found } \sigma \text{vz} \} \\ \text{find } \Gamma (\text{fs } i) \Leftarrow \text{case } \Gamma \{ \\ \text{find } (\text{bind } \Gamma \sigma) (\text{fs } i) \Leftarrow \text{view } (\text{find } \Gamma i) \{ \\ \text{find } (\text{bind } \Gamma \sigma) (\text{fs } (\text{fV } \tau i)) \Rightarrow \text{found } \tau (\text{vs } i) \} \} \} \} \} \right) \end{array}$$

5.1 Term and Terror

We can follow the same recipe for typechecking as we did for context lookup. Help me fill in the details:

Exercise 22 (fTm) *Implement the forgetful map:*

$$\underline{\text{let}} \left(\frac{\Gamma \text{ SCtxt } n; t : \text{STm } \Gamma \tau}{\text{fTm } \tau t : \text{RTm } n} \right)$$

But not every raw term is the forgetful image of a well typed term. We'll need

$$\underline{\text{data}} \left(\frac{\Gamma : \text{SCtxt } n}{\text{TError } \Gamma : \star} \right) \underline{\text{where}} \dots$$

$$\underline{\text{let}} \left(\frac{\Gamma : \text{SCtxt } n; e : \text{TError } \Gamma}{\text{fTError } e : \text{RTm } n} \right)$$

Exercise 24 (TError, fTError) *Fill out the definition of TError and implement fTError. (This will be easy, once you've done the missing exercise. The TErrors will jump out as we write the typechecker—they pack up the failure cases.)*

Let's start on the typechecking view. First, the checkability relation:

$$\underline{\text{data}} \left(\frac{\Gamma : \text{SCtxt } n; r : \text{RTm } n}{\text{Check } \Gamma r : \star} \right)$$

$$\underline{\text{where}} \left(\frac{t : \text{STm } \Gamma \tau}{\text{good } t : \text{Check } \Gamma (\text{fTm } \tau t)} \right); \left(\frac{e : \text{TError } \Gamma}{\text{bad } e : \text{Check } \Gamma (\text{fTError } e)} \right)$$

Next, let's start on the proof of checkability—sorry, the typechecker:

$$\underline{\text{let}} \left(\frac{}{\text{check } \Gamma r : \text{Check } \Gamma r} \right)$$

$$\text{check } \Gamma r \Leftarrow \underline{\text{rec}} r \{$$

$$\text{check } \Gamma r \Leftarrow \underline{\text{case}} r \{$$

$$\text{check } \Gamma (\text{rvar } i) \Leftarrow \underline{\text{view}} (\text{find } \Gamma i) \{$$

$$\text{check } \Gamma (\text{rvar } (\text{fV } \tau i)) \Rightarrow \text{good } (\text{svar } i) \}$$

$$\text{check } \Gamma (\text{rapp } f s) \Leftarrow \underline{\text{view}} (\text{check } \Gamma f) \{$$

$$\text{check } \Gamma (\text{rapp } (\text{fTm } \phi f) s) \Leftarrow \underline{\text{case}} \phi \{$$

$$\text{check } \Gamma (\text{rapp } (\text{fTm } \text{sNat } f) s) \Rightarrow \text{bad } \square$$

$$\text{check } \Gamma (\text{rapp } (\text{fTm } (\text{sFun } \sigma \tau) f) s) \Leftarrow \underline{\text{view}} (\text{check } \Gamma s) \{$$

$$\text{check } \Gamma (\text{rapp } (\text{fTm } (\text{sFun } \sigma \tau) f) (\text{fTm } \alpha s)) \square$$

$$\text{check } \Gamma (\text{rapp } (\text{fTm } (\text{sFun } \sigma \tau) f) (\text{fTError } e)) \Rightarrow \text{bad } \square \}}$$

$$\text{check } \Gamma (\text{rapp } (\text{fTError } e) s) \Rightarrow \text{bad } \square \}$$

$$\text{check } \Gamma (\text{rlda } \sigma t) \Leftarrow \underline{\text{view}} (\text{check } (\text{bind } \Gamma \sigma) t) \{$$

$$\text{check } \Gamma (\text{rlda } \sigma (\text{fTm } \tau t)) \Rightarrow \text{good } (\text{slda } t)$$

$$\text{check } \Gamma (\text{rlda } \sigma (\text{fTError } e)) \Rightarrow \text{bad } \square \}}$$

The story so far: we used **find** to check variables; we used **check** recursively to check the body of an **rlda** and packed up the successful outcome. Note that we don't need to write the types of the good terms—they're implicit in **STm**.

We also got some way with application: checking the function; checking that the function inhabits a function space; checking the argument. The only trouble is that our function expects a σ and we've got an α . We need to *see* if they're the same: that's the missing exercise.

Exercise 23 *Develop an equality view for **SType**:*

$$\begin{array}{l} \text{data} \left(\frac{\sigma, \tau : \text{SType}}{\text{Compare } \sigma \tau : \star} \right) \\ \text{where} \left(\frac{}{\text{same} : \text{Compare } \tau \tau} \right) ; \left(\frac{\sigma' : \text{Diff } \sigma}{\text{diff } \sigma' : \text{Compare } \sigma (\text{fDiff } \sigma \sigma')} \right) \\ \text{let} \left(\frac{}{\text{compare } \sigma \tau : \text{Compare } \sigma \tau : \star} \right) \end{array}$$

You'll need to define a representation of **STypes** which differ from a given σ and a forgetful map **fDiff** which forgets this difference.

How to go about it? Wait and see. Let's go back to application...

$$\begin{array}{l} \text{check } \Gamma (\text{rapp} (\text{fTm} (\text{sFun } \sigma \tau) f) (\text{fTm } \alpha s)) \Leftarrow \text{view} (\text{compare } \sigma \alpha) \{ \\ \text{check } \Gamma (\text{rapp} (\text{fTm} (\text{sFun } \sigma \tau) f) (\text{fTm } \sigma s)) \Rightarrow \text{good} (\text{sapp } f s) \\ \text{check } \Gamma (\text{rapp} (\text{fTm} (\text{sFun } \sigma \tau) f) (\text{fTm} (\text{fDiff } \sigma \sigma') s)) \Rightarrow \text{bad} [\] \} \end{array}$$

If we use your **compare** view, we can see directly that the types match in one case and mismatch in the other. For the former, we can now return a well typed application. The latter is definitely wrong.

We've done all the **good** cases, and we're left with choosing inhabitants of **TError** Γ for the **bad** cases. There's no reason why you shouldn't define **TError** Γ to make this as easy as possible. Just pack up the information which is lying around! For the case we've just seen, you could have:¹³

$$\left(\frac{\sigma' : \text{Diff } \sigma ; f : \text{STm } \Gamma (\text{sFun } \sigma \tau) ; s : \text{STm } \Gamma (\text{fDiff } \sigma \sigma')}{\text{mismatchError } \sigma' f s : \text{TError } \Gamma} \right)$$

$$\text{fTError} (\text{mismatchError } \sigma' f s) \Rightarrow \text{rapp} (\text{fTm } ? f) (\text{fTm } ? s)$$

This recipe gives one constructor for each **bad** case, and you don't have any choice about its declaration. There are two basic type errors—the above mismatch and the application of a non-function. The remaining three **bad** cases just propagate failure outwards: you get a type of located errors.

¹³ The ? means 'please infer'—it's often useful when writing forgetful maps. Why?

Of course, you'll need to develop **comparable** first. To define **Diff**, just play the same type-of-diagnostics game. Develop the equality test, much as you would with the Boolean version, but using the view recursively in order to see when the sources and targets of two **sFuns** are the same. If you need a hint, see [27].

What have we achieved? We've written a typechecker which not only returns *some* well typed term or error message, but, specifically, *the* well typed term or error message which corresponds to its input by **fTm** or **fTError**. That correspondance is directly expressed by a very high level derived form of pattern matching: not **rvar**, **rapp** or **rlda**, but 'well typed' or 'ill typed'.

5.2 A Typesafe and Total Interpreter

Once you have a well typed term, you can extract some operational benefit from its well-typedness—you can execute it without run-time checks. This example was inspired by Lennart Augustsson and Magnus Carlsson's interpreter for terms with a typing proof [5]. Epigram's inductive families allow us a more direct approach: we just write down a denotational semantics for well typed terms. Firstly, we must interpret **SType**:

$$\underline{\text{let}} \left(\frac{\tau : \mathbf{SType}}{\mathbf{Value} \tau : \star} \right) ; \begin{array}{l} \mathbf{Value} \tau \Leftarrow \underline{\text{rec}} \tau \{ \\ \mathbf{Value} \tau \Leftarrow \underline{\text{case}} \tau \{ \\ \mathbf{Value} \text{sNat} \Rightarrow \mathbf{Nat} \\ \mathbf{Value} (\text{sFun} \sigma \tau) \Rightarrow \mathbf{Value} \sigma \rightarrow \mathbf{Value} \tau \} \} \end{array}$$

Now we can explain how to interpret a context by an environment of values:

$$\underline{\text{data}} \left(\frac{\Gamma : \mathbf{SCtxt} \ n}{\mathbf{Env} \ \Gamma : \star} \right) \\ \underline{\text{where}} \left(\frac{}{\mathbf{empty} : \mathbf{Env} \ \mathbf{empty}} \right) ; \left(\frac{\gamma : \mathbf{Env} \ \Gamma ; v : \mathbf{Value} \ \sigma}{\mathbf{ebind} \ \gamma \ \sigma : \mathbf{Env} \ (\mathbf{bind} \ \Gamma \ \sigma)} \right)$$

Next, interpret variables by looking them up:

$$\underline{\text{let}} \left(\frac{\gamma : \mathbf{Env} \ \Gamma ; i : \mathbf{SVar} \ \Gamma \ \tau}{\mathbf{evar} \ \gamma \ i : \mathbf{Value} \ \tau} \right) ; \begin{array}{l} \mathbf{evar} \ \gamma \ i \Leftarrow \underline{\text{rec}} \ i \{ \\ \mathbf{evar} \ \gamma \ i \Leftarrow \underline{\text{case}} \ i \{ \\ \mathbf{evar} \ \gamma \ \mathbf{vz} \Leftarrow \underline{\text{case}} \ \gamma \{ \\ \mathbf{evar} \ (\mathbf{ebind} \ \gamma \ v) \ \mathbf{vz} \Rightarrow v \} \\ \mathbf{evar} \ \gamma \ (\mathbf{vs} \ i) \Leftarrow \underline{\text{case}} \ \gamma \{ \\ \mathbf{evar} \ (\mathbf{ebind} \ \gamma \ v) \ (\mathbf{vs} \ i) \Rightarrow \mathbf{evar} \ \gamma \ i \} \} \} \end{array}$$

Finally, interpret the well typed terms:

$$\underline{\text{let}} \left(\frac{\gamma : \text{Env } \Gamma ; t : \text{STm } \Gamma \tau}{\text{eval } \gamma t : \text{Value } \tau} \right)$$

$$\text{eval } \gamma t \Leftarrow \underline{\text{rec}} t \{$$

$$\text{eval } \gamma t \Leftarrow \underline{\text{case}} t \{$$

$$\text{eval } \gamma (\text{svar } i) \Rightarrow \text{eval } \gamma i$$

$$\text{eval } \gamma (\text{sapp } f s) \Rightarrow \text{eval } \gamma f (\text{eval } \gamma s)$$

$$\text{eval } \gamma (\text{slambda } t) \Rightarrow \lambda v \Rightarrow \text{eval } (\text{ebind } \gamma v) t \}}$$

Exercise 25 Make an environment whose entries are the constructors for `Nat`, together with some kind of iterator. Add two and two.

6 Epilogue

Well, we've learned to add two and two. It's true that Epigram is currently little more than a toy, but must it necessarily remain so? There is much work to do.

I hope I have shown that precise data structures can be manipulated successfully and in a highly articulate manner. You don't have to be content with giving orders to the computer and keeping your ideas to yourself. What has become practical is a notion of program as *effective explanation*, rather than merely an effective procedure. Upon what does this practicality depend?

- *adapting the programming language to suit dependent types*
Our conventional programming constructs are not well-suited either to cope with or to capitalize on the richness of dependent data structures. We have had to face up to the fact that inspecting one value can tell us more about types and about other values. And so it should: at long last, testing makes a difference! Moreover, the ability of types to talk about values gives us ready access to a new, more articulate way of programming with the high-level structure of values expressed directly as patterns.
- *using type information earlier in the programming process*
With so much structure—and computation—at the type level, keeping yourself type correct is inevitably more difficult. But it isn't necessary! Machines can check types and run programs, so use them! Interactive programming shortens the feedback loop, and it makes types a positive input to the programming process, not just a means to police its output.
- *changing the programs we choose to write*
We shouldn't expect dependently typed programming merely to extend the functional canon with new programs which could not be typed before. In order to exploit the power of dependent types to express and enforce stronger invariants, we need a new style of programming which explicitly *establishes* those invariants. We need to rework old programs, replacing uninformative types with informative ones.

6.1 Related Work

Epigram’s elder siblings are DML [39] and Cayenne [4]. DML equips ML programs with types refined by linear integer constraints and equips the typechecker with a constraint-solver. Correspondingly, many basic invariants, especially those involving sizes and ranges, can be statically enforced—this significantly reduces the overhead of run time checking [40]. Epigram has no specialist constraint-solver for arithmetic, although such a thing is a possible and useful extension. Epigram’s strength is in the diversity of its type-level language.

Cayenne is much more ambitious than DML and a lot closer to Epigram. It’s notorious for its looping typechecker, although (contrary to popular misconception) this is not an inevitable consequence of mixing dependent types with general recursion—recursion is implemented via fixpoints, so even structurally recursive programs can loop—you can always expand a fixpoint.

Cayenne’s main drawback is that it doesn’t support the kind of inductive families which Epigram inherited from the Alf system [21, 13]. It rules out those in which constructors only target *parts* of a family, the way `vnil` makes *empty* vectors and `vcons` makes *nonempty* vectors. This also rules out `SVar`, `STm` and all of our ‘views’. All of these examples can be given a cumbersome encoding if you are willing to work hard enough: I for one am not.

The Agda proof assistant [12], like Epigram, is very much in the spirit of Alf, but it currently imposes the same restrictions on inductive definitions as Cayenne and hence would struggle to support the programs in these notes—this unfortunate situation is unlikely to continue. Meanwhile Coq [11] certainly accepts the inductive definitions in this paper: it just has no practical support for programming with them—there is no good reason for this to remain so.

In fact, the closest programming language to Epigram at time of writing is Haskell, with `ghc`’s new ‘generalised algebraic data types’ [33]. These turn out to be, more or less, inductive families! Of course, in order to preserve the rigid separation of static types and dynamic terms, GADTs must be indexed by *type* expressions. It becomes quite easy to express examples like the type-safe interpreter, which exploit the invariants enforced by indexing. What is still far from obvious is how to *establish* invariants for run time data, as we did in our typechecker—this requires precisely the transfer of information from the dynamic to the static which is still excluded.

6.2 What is to be done?

We have only the very basic apparatus of dependently typed programming in place at the moment. We certainly need some way to analyse the results of *intermediate* computations in a way which reflects their significance for the existing type and value information—I have studiously avoided this issue in these notes. In [27], we propose a construct which adds the result of an intermediate computation to the collection of values being scrutinized on the left-hand side, at the same time abstracting it from types. This is not yet implemented.

We shall certainly need coinductive data in order to develop interactive systems. Inspired by the success of monads in Haskell, we shall also need to investigate the enhanced potential for ‘categorical packaging’ of programming in a language where the notion of category can be made abstract. And of course, there are all the ‘modern conveniences’: infix operators, ad-hoc polymorphism, generics, and so forth. These require design effort: the underlying expressivity is available, but we need good choices for their high-level presentation.

Work has already begun on a compiler for Epigram [10]: we have barely started to exploit our new wealth of static information for performance. We have the benefit of a large total fragment, in which evaluation strategy is unimportant and program transformation is no longer troubled by \perp . The fact that partial evaluation is already a fact of life for us must surely help also.

We need a library, but it’s not enough to import the standard presentation of standard functionality. Our library must support the idioms of dependently typed programming, which may well be different. Standardizing too early might be a mistake: we need to explore the design space for standard equipment.

But the greatest potential for change is in the tools of program development. Here, we have barely started. Refinement-style editing is great when you have a plan, but often we don’t. We need to develop *refactoring* technology for Epigram, so that we can sharpen our definitions as we learn from experiments. It’s seldom straight away that we happen upon exactly the indexed data structure we need.

Moreover, we need editing facilities that reflect the idioms of programming. Many data structures have a rationale behind them—they are intended to relate to other data structures in particular ways and support particular operations. At the moment we write none of this down. The well typed terms are supposed to be a more carefully indexed version of the raw terms—we should have been able to construct them explicitly as such. If only we could express our design principles then we could follow them deliberately. Currently, we engineer coincidences, dreaming up datatypes and operations as if from thin air.

But isn’t this just wishful thinking? I claim not. Dependent types, seen through the Curry-Howard lens, can characterize types and programs in a way which editing technology can exploit. We’ve already seen one class of logical principle reified as a programming operation—the \Leftarrow construct. We’ve been applying reasoning to the construction of programs on paper for years. We now have what we need to do the same effectively on a computer: a high-level programming language in which reasons and programs not merely coexist but coincide.

References

1. Andreas Abel and Thorsten Altenkirch. A predicative analysis of structural recursion. *Journal of Functional Programming*, 2000.
2. Thorsten Altenkirch and Bernhard Reus. Monadic presentations of lambda-terms using generalized inductive types. In *Computer Science Logic 1999*, 1999.
3. Lennart Augustsson. Compiling Pattern Matching. In Jean-Pierre Jouannaud, editor, *Functional Programming Languages and Computer Architecture*, volume 201 of *LNCS*, pages 368–381. Springer-Verlag, 1985.

4. Lennart Augustsson. Cayenne—a language with dependent types. In *ACM International Conference on Functional Programming '98*. ACM, 1998.
5. Lennart Augustsson and Magnus Carlsson. An exercise in dependent types: A well-typed interpreter. Available at <http://www.cs.chalmers.se/~augustss/cayenne/interp.ps>, 1999.
6. Françoise Bellegarde and James Hook. Substitution: A formal methods case study using monads and transformations. *Science of Computer Programming*, 1995.
7. Yves Bertot and Pierre Castéran. *Interactive Theorem Proving and Program Development, Coq'Art: The Calculus of Inductive Constructions*. Texts in Theoretical Computer Science. Springer-Verlag, 2004.
8. Richard Bird and Oege de Moor. *Algebra of Programming*. Prentice Hall, 1997.
9. Richard Bird and Ross Paterson. de Bruijn notation as a nested datatype. *Journal of Functional Programming*, 9(1):77–92, 1999.
10. Edwin Brady. *Practical Implementation of a Dependently Typed Functional Programming Language*. PhD thesis, University of Durham, 2005.
11. L'Équipe Coq. The Coq Proof Assistant Reference Manual. <http://pauillac.inria.fr/coq/doc/main.html>, 2001.
12. Catarina Coquand and Thierry Coquand. Structured Type Theory. In *Workshop on Logical Frameworks and Metalanguages*, 1999.
13. Thierry Coquand. Pattern Matching with Dependent Types. In Bengt Nordström, Kent Petersson, and Gordon Plotkin, editors, *Electronic Proceedings of the Third Annual BRA Workshop on Logical Frameworks (Båstad, Sweden)*, 1992.
14. Nicolas G. de Bruijn. Lambda Calculus notation with nameless dummies: a tool for automatic formula manipulation. *Indagationes Mathematicæ*, 34:381–392, 1972.
15. Peter Dybjer. Inductive Sets and Families in Martin-Löf's Type Theory. In Gérard Huet and Gordon Plotkin, editors, *Logical Frameworks*. CUP, 1991.
16. Daniel Fridlender and Mia Indrika. Do we need dependent types? *Journal of Functional Programming*, 10(4):409–415, 2000.
17. Eduardo Giménez. Codifying guarded definitions with recursive schemes. In Peter Dybjer, Bengt Nordström, and Jan Smith, editors, *Types for Proofs and Programs, '94*, volume 996 of *LNCS*, pages 39–59. Springer-Verlag, 1994.
18. Ralf Lämmel and Simon Peyton Jones. Scrap your boilerplate: a practical design pattern for generic programming. *ACM SIGPLAN Notices*, 38(3):26–37, March 2003. Proc. of the ACM SIGPLAN Workshop on Types in Language Design and Implementation (TLDI 2003).
19. Zhaohui Luo. *Computation and Reasoning: A Type Theory for Computer Science*. Oxford University Press, 1994.
20. Zhaohui Luo and Robert Pollack. LEGO Proof Development System: User's Manual. Technical Report ECS-LFCS-92-211, Laboratory for Foundations of Computer Science, University of Edinburgh, 1992.
21. Lena Magnusson and Bengt Nordström. The ALF proof editor and its proof engine. In Henk Barendregt and Tobias Nipkow, editors, *Types for Proofs and Programs, LNCS 806*. Springer-Verlag, 1994. Selected papers from the Int. Workshop TYPES '93, Nijmegen, May 1993.
22. Per Martin-Löf. A theory of types. manuscript, 1971.
23. Conor McBride. *Dependently Typed Functional Programs and their Proofs*. PhD thesis, University of Edinburgh, 1999. Available from <http://www.lfcs.informatics.ed.ac.uk/reports/00/ECS-LFCS-00-419/>.
24. Conor McBride. Faking It (Simulating Dependent Types in Haskell). *Journal of Functional Programming*, 12(4& 5):375–392, 2002. Special Issue on Haskell.

25. Conor McBride. First-Order Unification by Structural Recursion. *Journal of Functional Programming*, 13(6), 2003.
26. Conor McBride. Epigram, 2004. <http://www.dur.ac.uk/CARG/epigram>.
27. Conor McBride and James McKinna. The view from the left. *Journal of Functional Programming*, 14(1), 2004.
28. Fred McBride. *Computer Aided Manipulation of Symbols*. PhD thesis, Queen's University of Belfast, 1970.
29. Robin Milner, Mads Tofte, Robert Harper, and David MacQueen. *The Definition of Standard ML, revised edition*. MIT Press, 1997.
30. Chris Okasaki. *Purely Functional Data Structures*. Cambridge University Press, 1998.
31. Chris Okasaki. From Fast Exponentiation to Square Matrices: An Adventure in Types. In *ACM International Conference on Functional Programming '99*, 1999.
32. Simon Peyton Jones and John Hughes, editors. *Haskell'98: A Non-Strict Functional Language*, 1999. Available from <http://www.haskell.org/definition>.
33. Simon Peyton Jones, Geoffrey Washburn, and Stephanie Weirich. Wobbly types: type inference for generalised algebraic data types. Unpublished, 2004.
34. Dag Prawitz. *Natural Deduction—A proof theoretical study*. Almqvist and Wiksell, Stockholm, 1965.
35. Bengt Nordström, Kent Petersson, and Jan Smith. *Programming in Martin-Löf's type theory: an introduction*. Oxford University Press, 1990.
36. Alan Robinson. A Machine-oriented Logic Based on the Resolution Principle. *Journal of the ACM*, 12:23–41, 1965.
37. David Turner. Elementary Strong Functional Programming. In *Functional Programming Languages in Education, First International Symposium*, volume 1022 of *LNCS*. Springer-Verlag, 1995.
38. Philip Wadler. Views: A way for pattern matching to cohabit with data abstraction. In *Proceedings of POPL '87*. ACM, 1987.
39. Hongwei Xi. *Dependent Types in Practical Programming*. PhD thesis, Department of Mathematical Sciences, Carnegie Mellon University, 1998.
40. Hongwei Xi and Frank Pfenning. Eliminating array bound checking through dependent types. In *Proceedings of the Conference on Programming Language Design and Implementation (PLDI'98)*. ACM Press, 1998.